
Chapter 17. Socket Programming
728

1249985
This chapter shows how to use sockets for programming network clients and servers.
Advanced I/O techniques for sockets are described, including nonblocking I/O and control
over I/O uffering. Tcl commands discussed are: socket, fconfigure, and
http::geturl.
1249985
Sockets are network communication channels. The sockets described in this chapter use the
TCP network protocol, although you can find Tcl extensions that create sockets using other
protocols. TCP provides a reliable byte stream between two hosts connected to a network.
TCP handles all the issues about routing information across the network, and it automatically
recovers if data is lost or corrupted along the way. TCP is the basis for other protocols like
Telnet, FTP, and HTTP.
1249985
A Tcl script can use a network socket just like an open file or pipeline. Instead of using the
Tcl open command, you use the socket command to open a socket. Then you use gets,
puts, and read to transfer data. The close command closes a network socket.
1249985
Network programming distinguishes between clients and servers. A server is a process or
program that runs for long periods of time and controls access to some resource. For example,
an FTP server governs access to files, and an HTTP server provides access to hypertext pages
on the World Wide Web. A client typically connects to the server for a limited time in order
to gain access to the resource. For example, when a Web browser fetches a hypertext page,
it is acting as a client. The extended examples in this chapter show how to program the client
side of the HTTP protocol.
1249985

Networking Extensions for Tcl
621961
This chapter describes the basic programming techniques for sockets. Socket programing
in Tcl is pretty easy, and a variety of extensions have been created to handle common
protocols. This section reviews some of the packages that are available, and then the rest of
the chapter describes how to program sockets yourself.
1249985

Scotty
628024

The Scotty extension supports many network protocols.
1249985

The Scotty Tcl extension provides access to other network protocols like UDP, DNS, and RPC.
It also supports the SNMP network management protocol and the MIB database associated
with SNMP. Scotty is a great extension package that is widely used for network management
applications. It is a C-level extension, so you have to compile it yourself or find a binary
distribution. Its home page is:
1249985

http://wwwsnmp.cs.utwente.nl/~schoenw/scotty/

Standard Tcl Library
628024
The Standard Tcl Library (tcllib) has several packages that support widely used TCP-based
protocols. These are all pure-Tcl implementations. There are packages for:
1249985

• DNS client. Map between hostnames and IP addresses.

• FTP client. Open FTP connections and download files from FTP servers.

• FTP server. Implement a simple, extensible FTP server.

• IRC client. Implement a chat client.

• NNTP client. Fetch news from a news server.

• POP3 client. Post Office Protocol lets you fetch email from mail servers.

• POP3 server. Implement a mail server.

• SMTP client. Send email via the SMTP protocol.

• SMTP server. Accept incoming email via SMTP.

• URI manipulation. Package for parsing URLs.
There is good on-line documentation for these packages at:

http://tcllib.sourceforge.net/tcllib/doc/

HTTP
628024
The Tcl distribution includes an HTTP client, which is described on page 251. You don't need
to add tcllib to get this. In addition, there is a nice web server built in Tcl, which is the topic
of Chapter 18.
1249985

Client Sockets
621961

1249985
A client opens a socket by specifying the host address and port number for the server of the
socket. The host address gives the network location (i.e., which computer), and the port
selects a particular server from all the possible servers that may be running on that host. For
example, HTTP servers typically use port 80, while FTP servers use port 20. The following
example shows how to open a client socket to a Web server:
1249985

set s [socket www.tcl.tk 80]

There are two forms for host names. The previous example uses a domain name:
www.tcl.tk. You can also specify raw IP addresses, which are specified with four dot-
separated integers (e.g., 192.220.75.86). A domain name is mapped into a raw IP address by
the system software, and it is almost always a better idea to use a domain name in case the
IP address assignment for the host changes. This can happen when hosts are upgraded or
they move to a different part of the network.
1249985
Some systems also provide symbolic names for well-known port numbers. For example,
instead of using 20 for the FTP service, you can use ftp. On UNIX systems, the well-known
port numbers are listed in the file named /etc/services.
1249985

Client Socket Options
628024
The socket command accepts some optional arguments when opening the client-side
socket. The general form of the command is:
1249985

socket ?-async? ?-myaddr 
address
? ?-myport 
myport
? 
host port

            

Ordinarily the address and port on the client side are chosen automatically. If your computer
has multiple network interfaces, you can select one with the -myaddr option. The
address value can be a domain name or an IP address. If your application needs a specific
client port, it can choose one with the -myport option. If the port is in use, the socket
command will raise an error.
1249985
The -async option causes connection to happen in the background, and the socket
command returns immediately. The socket becomes writable when the connection
completes, or fails. You can use fileevent to get a callback when this occurs. This is shown
in Example 17-1. If you use the socket before the connection completes, and the socket is in
blocking mode, then Tcl automatically blocks and waits for the connection to complete. If
the socket is in nonblocking mode, attempts to use the socket return immediately. The
gets and read commands would return -1, and fblocked would return 1 in this situation.
1249985
In some cases, it can take a long time to open the connection to the server. Usually this occurs
when the server host is down, and it may take longer than you want for the connection to
time out. The following example sets up a timer with after so that you can choose your
own timeout limit on the connection:
1249985

Example 17-1. Opening a client socket with a timeout

proc Socket_Client {host port timeout} {
   global connected
   after $timeout {set connected timeout}
   set sock [socket -async $host $port]
   fileevent $sock w {set connected ok}
   vwait connected
   fileevent $sock w {}
   if {$connected == "timeout"} {
      return -code error timeout
   } else {
      return $sock

   }
}

Server Sockets
621961

1249985
A TCP server socket allows multiple clients. The way this works is that the socket command
creates a listening socket, and then new sockets are created when clients make connections
to the server. Tcl takes care of all the details and makes this easy to use. You simply specify
a port number and give the socket command a callback to execute when a client connects
to your server socket. The callback is just a Tcl command. A simple example is shown below:

1249985

Example 17-2. Opening a server socket

set listenSocket [socket -server Accept 2540]
proc Accept {newSock addr port} {
   puts "Accepted $newSock from $addr port $port"
}
vwait forever

The Accept command is the callback made when clients connect to the server. Tcl adds
additional arguments to the callback before it calls it. The arguments are the new socket
connection, and the host and port number of the remote client. In this simple example,
Accept just prints out its arguments.
1249985
The vwait command puts Tcl into its event loop so that it can do the background processing
necessary to accept connections. The vwait command will wait until the forever variable
is modified, which won't happen in this simple example. The key point is that Tcl processes
other events (e.g., network connections and other file I/O) while it waits. If you have a Tk
application (e.g., wish), then it already has an event loop to handle window system events,
so you do not need to use vwait. The Tcl event loop is discussed on page 227.
1249985

Server Socket Options
628024
By default, Tcl lets the operating system choose the network interface used for the server
socket, and you simply supply the port number. If your computer has multiple interfaces,

you may want to specify a particular one. Use the -myaddr option for this. The general form
of the command to open server sockets is:
1249985

socket -server 
callback
 ?-myaddr 
address
? 
port

            

The last argument to the socket command is the server's port number. For your own
unofficial servers, you'll need to pick port numbers higher than 1024 to avoid conflicts with
existing services. UNIX systems prevent user programs from opening server sockets with port
numbers less than 1024. If you use 0 as the port number, then the operating system will pick
the listening port number for you. You must use fconfigure to find out what port you
have:
1249985

fconfigure $sock -sockname
=> ipaddr hostname port

            

The Echo Service
621961

1249985

Example 17-3. The echo service

proc Echo_Server {port} {
   global echo
   set echo(main) [socket -server EchoAccept $port]
}
proc EchoAccept {sock addr port} {
   global echo
   puts "Accept $sock from $addr port $port"
   set echo(addr,$sock) [list $addr $port]
   fconfigure $sock -buffering line
   fileevent $sock readable [list Echo $sock]
}
proc Echo {sock} {
   global echo
   if {[eof $sock] || [catch {gets $sock line}]} {
      # end of file or abnormal connection drop
      close $sock
      puts "Close $echo(addr,$sock)"
      unset echo(addr,$sock)
   } else {

      if {[string compare $line "quit"] == 0} {
         # Prevent new connections.
         # Existing connections stay open.
         close $echo(main)
      }
      puts $sock $line
   }
}

The echo server accepts connections from clients. It reads data from the clients and writes
that data back. The example uses fileevent to wait for data from the client, and it uses
fconfigure to adjust the buffering behavior of the network socket. You can use Example
17-3 as a template for more interesting services.
1249985
The Echo_Server procedure opens the socket and saves the result in echo(main). When
this socket is closed later, the server stops accepting new connections but existing
connections won't be affected. If you want to experiment with this server, start it and wait
for connections like this:
1249985

Echo_Server 2540
vwait forever

The EchoAccept procedure uses the fconfigure command to set up line buffering. This
means that each puts by the server results in a network transmission to the client. The
importance of this will be described in more detail later. A complete description of the
fconfigure command is given in Chapter 16. The EchoAccept procedure uses the
fileevent command to register a procedure that handles I/O on the socket. In this
example, the Echo procedure will be called whenever the socket is readable. Note that it is
not necessary to put the socket into nonblocking mode when using the fileevent callback.
The effects of nonblocking mode are discussed on page 232.
1249985
EchoAccept saves information about each client in the echo array. This is used only to print
out a message when a client closes its connection. In a more sophisticated server, however,
you may need to keep more interesting state about each client. The name of the socket
provides a convenient handle on the client. In this case, it is used as part of the array index.
1249985
The Echo procedure first checks to see whether the socket has been closed by the client or
there is an error when reading the socket. The if expression only performs the gets if the
eof does not return true:
1249985

if {[eof $sock] || [catch {gets $sock line}]} {

Closing the socket automatically clears the fileevent registration. If you forget to close
the socket upon the end of file condition, the Tcl event loop will invoke your callback
repeatedly. It is important to close it when you detect end of file.

1249985

Example 17-4. A client of the echo service

proc Echo_Client {host port} {
   set s [socket $host $port]
   fconfigure $s -buffering line
   return $s
}
set s [Echo_Client localhost 2540]
puts $s "Hello!"
gets $s
=> Hello!

                        

In the normal case, the server simply reads a line with gets and then writes it back to the
client with puts. If the line is "quit," then the server closes its main socket. This prevents any
more connections by new clients, but it doesn't affect any clients that are already connected.
1249985
Example 17-4 shows a sample client of the Echo service. The main point is to ensure that the
socket is line buffered so that each puts by the client results in a network transmission. (Or,
more precisely, each newline character results in a network transmission.) If you forget to set
line buffering with fconfigure, the client's gets command will probably hang because
the server will not get any data; it will be stuck in buffers on the client.
1249985

Fetching a URL with HTTP
621961

1249985
The HyperText Transport Protocol (HTTP) is the protocol used on the World Wide Web. This
section presents a procedure to fetch pages or images from a server on the Web. Items in
the Web are identified with a Universal Resource Location (URL) that specifies a host, port,
and location on the host. The basic outline of HTTP is that a client sends a URL to a server,
and the server responds with some header information and some content data. The header
information describes the content, which can be hypertext, images, postscript, and more.
1249985

Example 17-5. Opening a connection to an HTTP server

proc Http_Open {url} {
   global http
   if {![regexp -nocase {^(http://)?([^:/]+)(:([0-9]+))?(/.*)} \
          $url x protocol server y port path]} {
      error "bogus URL: $url"
   }

   if {[string length $port] == 0} {
      set port 80
   }
   set sock [socket $server $port]
   puts $sock "GET $path HTTP/1.0"
   puts $sock "Host: $server"
   puts $sock "User-Agent: Tcl/Tk Http_Open"
   puts $sock ""
   flush $sock
   return $sock
}

The Http_Open procedure uses regexp to pick out the server and port from the URL. This
regular expression is described in detail on page 159. The leading http:// is optional, and
so is the port number. If the port is left off, then the standard port 80 is used. If the regular
expression matches, then a socket command opens the network connection.
1249985
The protocol begins with the client sending a line that identifies the command (GET), the
path, and the protocol version. The path is the part of the URL after the server and port
specification. The rest of the request is lines in the following format:
1249985

                
key: value

            

The Host identifies the server, which supports servers that implement more than one server
name. The User-Agent identifies the client program, which is often a browser like Netscape
Navigator, Mozilla, or Internet Explorer. The key-value lines are terminated with a blank line.
This data is flushed out of the Tcl buffering system with the flush command. The server will
respond by sending the URL contents back over the socket. This is described shortly, but first
we consider proxies.
1249985

Proxy Servers
628024
A proxy is used to get through firewalls that many organizations set up to isolate their network
from the Internet. The proxy accepts HTTP requests from clients inside the firewall and then
forwards the requests outside the firewall. It also relays the server's response back to the
client. The protocol is nearly the same when using the proxy. The difference is that the
complete URL is passed to the GET command so that the proxy can locate the server. Example
17-6 uses a proxy if one is defined:
1249985

Example 17-6. Opening a connection through a HTTP proxy

# Http_Proxy sets or queries the proxy
proc Http_Proxy {{new {}}} {
   global http
   if ![info exists http(proxy)] {
      return {}
   }
   if {[string length $new] == 0} {
      return $http(proxy):$http(proxyPort)
   } else {
      regexp {^([^:]+):([0-9]+)$} $new x \
         http(proxy) http(proxyPort)
   }
}

proc Http_Open {url {cmd GET} {query {}}} {
   global http
   if {![regexp -nocase {^(http://)?([^:/]+)(:([0-9]+))?(/.*)} \
          $url x protocol server y port path]} {
      error "bogus URL: $url"
   }
   if {[string length $port] == 0} {
      set port 80
   }
   if {[info exists http(proxy)] &&
          [string length $http(proxy)]} {
      set sock [socket $http(proxy) $http(proxyPort)]
      puts $sock "$cmd http://$server:$port$path HTTP/1.0"
   } else {
      set sock [socket $server $port]
      puts $sock "$cmd $path HTTP/1.0"
   }
   puts $sock "User-Agent: Tcl/Tk Http_Open"
   puts $sock "Host: $server"
   if {[string length $query] > 0} {
      puts $sock "Content-Length: [string length $query]"
      puts $sock ""
      puts $sock $query
   }
   puts $sock ""
   flush $sock
   fconfigure $sock -blocking 0
   return $sock
}

The HEAD Request
628024
In Example 17-6, the Http_Open procedure takes a cmd parameter so that the user of
Http_Open can perform different operations. The GET operation fetches the contents of a
URL. The HEAD operation just fetches the description of a URL, which is useful to validate a
URL. The POST operation transmits query data to the server (e.g., values from a form) and
also fetches the contents of the URL. All of these operations follow a similar protocol. The
reply from the server is a status line followed by lines that have key-value pairs. This format
is similar to the client's request. The reply header is followed by content data with GET and
POST operations. Example 17-7 implements the HEAD command, which does not involve
any reply data:
1249985

Example 17-7. Http_Head validates a URL

proc Http_Head {url} {
   upvar #0 $url state
   catch {unset state}
   set state(sock) [Http_Open $url HEAD]
   fileevent $state(sock) readable [list HttpHeader $url]
   # Specify the real name, not the upvar alias, to vwait
   vwait $url\(status)
   catch {close $state(sock)}
   return $state(status)
}
proc HttpHeader {url} {
   upvar #0 $url state
   if {[eof $state(sock)]} {
      set state(status) eof
      close $state(sock)
      return
   }
   if {[catch {gets $state(sock) line} nbytes]} {
      set state(status) error
      lappend state(headers) [list error $nbytes]
      close $state(sock)
      return
   }
   if {$nbytes < 0} {
      # Read would block
      return
   } elseif {$nbytes == 0} {
      # Header complete
      set state(status) head
   } elseif {![info exists state(headers)]} {
      # Initial status reply from the server
      set state(headers) [list http $line]
   } else {
      # Process key-value pairs
      regexp {^([^:]+): *(.*)$} $line x key value
      lappend state(headers) [string tolower $key] $value
   }
}

The Http_Head procedure uses Http_Open to contact the server. The HttpHeader
procedure is registered as a fileevent handler to read the server's reply. A global array
keeps state about each operation. The URL is used in the array name, and upvar is used to
create an alias to the name (upvar is described on page 92):
1249985

upvar #0 $url state

You cannot use the upvar alias as the variable specified to vwait. Instead, you must use
the actual name. The backslash turns off the array reference in order to pass the name of the
array element to vwait, otherwise Tcl tries to reference url as an array:
1249985

vwait $url\(status)

The HttpHeader procedure checks for special cases: end of file, an error on the gets, or a
short read on a nonblocking socket. The very first reply line contains a status code from the
server that is in a different format than the rest of the header lines:

1249985

                
code message

            

The code is a three-digit numeric code. 200 is OK. Codes in the 400's and 500's indicate an
error. The codes are explained fully in RFC 1945 that specifies HTTP 1.0. The first line is saved
with the key http:
1249985

set state(headers) [list http $line]

The rest of the header lines are parsed into key-value pairs and appended onto state
(headers). This format can be used to initialize an array:
1249985

array set header $state(headers)

When HttpHeader gets an empty line, the header is complete and it sets the state
(status) variable, which signals Http_Head. Finally, Http_Head returns the status to its
caller. The complete information about the request is still in the global array named by the
URL. Example 17-8 illustrates the use of Http_Head:
1249985

Example 17-8. Using Http_Head

set url http://www.sun.com/
set status [Http_Head $url]
=> eof

upvar #0 $url state
array set info $state(headers)
parray info
info(http)          HTTP/1.0 200 OK

                            
info(server)        Apache/1.1.1

                            
info(last-modified) Nov ...

                            
info(content-type)  text/html

                        

The GET and POST Requests
628024
Example 17-9 shows Http_Get, which implements the GET and POST requests. The
difference between these is that POST sends query data to the server after the request

header. Both operations get a reply from the server that is divided into a descriptive header
and the content data. The Http_Open procedure sends the request and the query, if present,
and reads the reply header. Http_Get reads the content.
1249985
The descriptive header returned by the server is in the same format as the client's request.
One of the key-value pairs returned by the server specifies the Content-Type of the URL.
The content-types come from the MIME standard, which is described in RFC 1521. Typical
content-types are:
1249985

• text/html — HyperText Markup Language (HTML), which is introduced in Chapter
3.

• text/plain — plain text with no markup.
• image/gif — image data in GIF format.
• image/jpeg — image data in JPEG format.
• application/postscript — a postscript document.
• application/x-tcl — a Tcl program! This type is discussed in Chapter 20.

Example 17-9. Http_Get fetches the contents of a URL

proc Http_Get {url {query {}}} {
   upvar #0 $url state        ;# Alias to global array
   catch {unset state}        ;# Aliases still valid.
   if {[string length $query] > 0} {
      set state(sock) [Http_Open $url POST $query]
   } else {
      set state(sock) [Http_Open $url GET]
   }
   set sock $state(sock)
   fileevent $sock readable [list HttpHeader $url]

   # Specify the real name, not the upvar alias, to vwait
   vwait $url\(status)
   set header(content-type) {}
   set header(http) "500 unknown error"
   array set header $state(headers)

   # Check return status.
   # 200 is OK, other codes indicate a problem.
   regsub "HTTP/1.. " $header(http) {} header(http)
   if {![string match 2* $header(http)]} {
      catch {close $sock}
      if {[info exists header(location)] &&
            [string match 3* $header(http)]} {
         # 3xx is a redirection to another URL
         set state(link) $header(location)
         return [Http_Get $header(location) $query]
      }
      return -code error $header(http)
   }
   # Set up to read the content data
   switch -glob -- $header(content-type) {
      text/*    {
         # Read HTML into memory
         fileevent $sock readable [list HttpGetText $url]
      }

      default   {
         # Copy content data to a file
         fconfigure $sock -translation binary
         set state(filename) [File_TempName http]
         if [catch {open $state(filename) w} out] {
            set state(status) error
            set state(error) $out
            close $sock
            return $header(content-type)
         }
         set state(fd) $out
         fcopy $sock $out -command [list HttpCopyDone $url]
      }
   }
   vwait $url\(status)
   return $header(content-type)
}

Http_Get uses Http_Open to initiate the request, and then it looks for errors. It handles
redirection errors that occur if a URL has changed. These have error codes that begin with
3. A common case of this error is when a user omits the trailing slash on a URL (e.g., http://
www.tcl.tk). Most servers respond with:
1249985

302 Document has moved
Location: http://www.tcl.tk/

If the content-type is text, then Http_Get sets up a fileevent handler to read this
data into memory. The socket is in nonblocking mode, so the read handler can read as much
data as possible each time it is called. This is more efficient than using gets to read a line at
a time. The text will be stored in the state(body) variable for use by the caller of
Http_Get. Example 17-10 shows the HttpGetText fileevent handler:
1249985

Example 17-10. HttpGetText reads text URLs

proc HttpGetText {url} {
   upvar #0 $url state
   if {[eof $state(sock)]} {
      # Content complete
      set state(status) done
      close $state(sock)
   } elseif {[catch {read $state(sock)} block]} {
      set state(status) error
      lappend state(headers) [list error $block]
      close $state(sock)
   } else {
      append state(body) $block
   }
}

The content may be in binary format. This poses a problem for Tcl 7.6 and earlier. A null
character will terminate the value, so values with embedded nulls cannot be processed safely
by Tcl scripts. Tcl 8.0 supports strings and variable values with arbitrary binary data. Example
17-9 uses fcopy to copy data from the socket to a file without storing it in Tcl variables. This

command was introduced in Tcl 7.5 as unsupported0, and became fcopy in Tcl 8.0. It
takes a callback argument that is invoked when the copy is complete. The callback gets
additional arguments that are the bytes transferred and an optional error string. In this case,
these arguments are added to the url argument specified in the fcopy command. Example
17-11 shows the HttpCopyDone callback:
1249985

Example 17-11. HttpCopyDone is used with fcopy

proc HttpCopyDone {url bytes {error {}}} {
   upvar #0 $url state
   if {[string length $error]} {
      set state(status) error
      lappend state(headers) [list error $error]
   } else {
      set state(status) ok
   }
   close $state(sock)
   close $state(fd)
}

The user of Http_Get uses the information in the state array to determine the status of
the fetch and where to find the content. There are four cases to deal with:
1249985

• There was an error, which is indicated by the state(error) element.

• There was a redirection, in which case, the new URL is in state(link). The client of Http_Get should change the URL and
look at its state instead. You can use upvar to redefine the alias for the state array:

upvar #0 $state(link) state
• There was text content. The content is in state(body).

• There was another content-type that was copied to state(filename).

The fcopy Command
628024
The fcopy command can do a complete copy in the background. It automatically sets up
fileevent handlers, so you do not have to use fileevent yourself. It also manages its
buffers efficiently. The general form of the command is:
1249985

fcopy 
input output
 ?-size 
size
? ?-command 
callback
?

The -command argument makes fcopy work in the background. When the copy is complete
or an error occurs, the callback is invoked with one or two additional arguments: the
number of bytes copied, and, in the case of an error, it is also passed an error string:
1249985

fcopy $in $out -command [list CopyDone $in $out]
proc CopyDone {in out bytes {error {}} {
     close $in ; close $out
}

With a background copy, the fcopy command transfers data from input until end of file
or size bytes have been transferred. If no -size argument is given, then the copy goes
until end of file. It is not safe to do other I/O operations with input or output during a
background fcopy. If either input or output gets closed while the copy is in progress, the
current copy is stopped. If the input is closed, then all data already queued for output is
written out.
1249985
Without a -command argument, the fcopy command reads as much as possible depending
on the blocking mode of input and the optional size parameter. Everything it reads is
queued for output before fcopy returns. If output is blocking, then fcopy returns after
the data is written out. If input is blocking, then fcopy can block attempting to read
size bytes or until end of file.
1249985
The fcopy command had a bug which ignored the encoding on the channels which was
corrected in 8.3.4.
1249985

The http Package
621961
The standard Tcl library includes an http package that is based on the code I wrote for this
chapter. This section documents the package, which has a slightly different interface. The
library version uses namespaces and combines the Http_Get, Http_Head, and
Http_Post procedures into a single http::geturl procedure. The examples in this
chapter are still interesting, but you should use the standard http package for your
production code.
1249985

http::config
628024
The http::config command is used to set the proxy information, time-outs, and the
User-Agent and Accept headers that are generated in the HTTP request. You can specify

the proxy host and port, or you can specify a Tcl command that is run to determine the proxy.
With no arguments, http::config returns the current settings:
1249985

http::config
=> -accept */* -proxyfilter http::ProxyRequired

                
-proxyhost {}
 -
proxyport {}

                
-useragent {Tcl http client package 2.4}

            

If you specify just one option, its value is returned:
1249985

http::config -proxyfilter
=> http::ProxyRequired

            

You can set one or more options:
1249985

http::config -proxyhost webcache.eng -proxyport 8080

The default proxy filter just returns the -proxyhost and -proxyport values if they are
set. You can supply a smarter filter that picks a proxy based on the host in the URL. The proxy
filter is called with the hostname and should return a list of two elements, the proxy host and
port. If no proxy is required, return an empty list.
1249985

http::geturl
628024
The http::geturl procedure does a GET, POST, or HEAD transaction depending on its
arguments. By default, http::geturl blocks until the request completes and it returns a
token that represents the transaction. As described below, you use the token to get the results
of the transaction. If you supply a -command callback option, then http::geturl
returns immediately and invokes callback when the transaction completes. The callback
is passed the token that represents the transaction.
1249985
For simple applications you can simply block on the transaction:
1249985

set token [http::geturl www.beedub.com/index.html]

=> http::1

            

The leading http:// in the URL is optional. The return value is a token that represents the
transaction. There are other http:: commands that return information when passed the
token. The token is also the name of an array that contains state about the transaction. Make
sure to clean up this array to free memory when you are done:
1249985

http::cleanup $token

If you need to access the array directly, use upvar to create an alias:
1249985

upvar #0 $token data

Table 17-1 lists the options to http::geturl.
1249985

Table 17-1. Options to the http::geturl command

-binary boolean
1249985

Specifies whether we should do a binary transfer of the
data. (Tcl 8.3)
1249985

-blocksize num
1249985

Block size when copying to a channel.
1249985

-channel fileID
1249985

The fileID is an open file or socket. The URL data is
copied to this channel instead of saving it in memory.
1249985

-command callback
1249985

Calls callback when the transaction completes. The
token from http::geturl is passed to callback.
1249985

-handler command
1249985

Called from the event handler to read data from the URL.
1249985

-headers list
1249985

The list specifies a set of headers that are included in the
HTTP request. The list alternates between header keys and
values.
1249985

-progress command
1249985

Calls command after each block is copied to a channel. It
gets called with three parameters:
1249985

command token totalsize currentsize
1249985

-query codedstring
1249985

Issues a POST request with the codedstring form data.
1249985

-queryblocksize num
1249985

Block size when copying to the query channel.
1249985

-querychannel fileID
1249985

The fileID is an open file or socket. The query data is
copied from this channel instead of passed in a string.
1249985

-queryprogress command
1249985

Calls command after each block is copied from the query
channel. It gets called with three parameters:
1249985
command token totalsize currentsize
1249985

-timeout msec
1249985

Aborts the request after msec milliseconds have elapsed.
1249985

-type mime-type
1249985

Use mime-type as the Content-Type value during a POST
operation.
1249985

-validate bool
1249985

If bool is true, a HEAD request is made.
1249985

Table 17-2 lists the access functions to the state array.
1249985

Table 17-2. The http support procedures

http::cleanup $token
1249985

Unsets the state array named by $token.
1249985

http::code $token
1249985

Returns state(http).
1249985

http::data $token
1249985

Returns state(body).
1249985

http::error $token
1249985

Returns state(error).
1249985

http::ncode $token
1249985

Returns the numeric return code contained in state
(http).
1249985

http::size $token
1249985

Return the number of bytes read from the URL so far.
1249985

http::status $token
1249985

Returns state(status).
1249985

http::wait $token
1249985

Blocks until the transaction completes.
1249985

The array elements are listed in Table 17-3:
1249985

Table 17-3. Elements of the http::geturl state array

body
1249985

The contents of the URL.
1249985

charset
1249985

The value of the charset attribute from the Content-Type meta-data value.
If none was specified, this defaults to the RFC standard iso8859-1.
1249985

coding
1249985

A copy of the Content-Encoding meta-data value.
1249985

currentsize
1249985

The current number of bytes transferred.
1249985

error
1249985

An explanation of why the transaction was aborted.
1249985

http
1249985

The HTTP reply status.
1249985

meta
1249985

A list of the keys and values in the reply header.
1249985

posterror
1249985

An explanation of why the transaction was aborted when writing post
query data, if any.
1249985

status
1249985

The current status: pending, ok, eof, or reset.
1249985

totalsize
1249985

The expected size of the returned data.
1249985

type
1249985

The content type of the returned data.
1249985

url
1249985

The URL of the request.
1249985

You can take advantage of the asynchronous interface by specifying a command that is called
when the transaction completes. The callback is passed the token returned from
http::geturl so that it can access the transaction state:
1249985

http::geturl $url -command [list Url_Display $text $url]
proc Url_Display {text url token} {
   upvar #0 $token state
   # Display the url in text
}

You can have http::geturl copy the URL to a file or socket with the -channel option.
This is useful for downloading large files or images. In this case, you can get a progress callback
so that you can provide user feedback during the transaction. Example 17-12 shows a simple
downloading script:
1249985

Example 17-12. Downloading files with http::geturl

#!/usr/local/bin/tclsh8.4
if {$argc < 2} {
   puts stderr "Usage: $argv0 url file"
   exit 1
}
package require http
set url [lindex $argv 0]
set file [lindex $argv 1]
set out [open $file w]
proc progress {token total current} {
   puts -nonewline "."
}
http::config -proxyhost webcache.eng -proxyport 8080
set token [http::geturl $url -progress progress \
   -headers {Pragma no-cache} -channel $out]
close $out
# Print out the return header information
puts ""
upvar #0 $token state
puts $state(http)
foreach {key value} $state(meta) {
   puts "$key: $value"
}
exit 0

http::formatQuery
628024
If you specify form data with the -query option, then http::geturl does a POST
transaction. You need to encode the form data for safe transmission. The
http::formatQuery procedure takes a list of keys and values and encodes them in x-
www-url-encoded format. Pass this result as the query data:
1249985

http::formatQuery name "Brent Welch" title "Tcl Programmer"
=> name=Brent+Welch&title=Tcl+Programmer

            

http::register and http::unregister
628024
The http::register procedure registers a protocol handler for URL protocols other than
HTTP. The http::unregister procedure removes the handler registration. The primary
application is to provide secure web access via HTTPS and the TLS extension.
1249985

package require tls
http::register https 443 ::tls::socket
set token [http::geturl https://my.secure.site/]

http::reset
628024
You can cancel an outstanding transaction with http::reset:
1249985

http::reset $token

This is done automatically when you setup a -timeout with http::config.
1249985

http::cleanup
628024
When you are done with the data returned from http::geturl, use the
http::cleanup procedure to unset the state variable used to store the data.
1249985

Basic Authentication
621961
Web pages are often password protected. The most common form of this uses a protocol
called Basic Authentication, which is not very strong, but easy to implement. With this
scheme, the server responds to an HTTP request with a 401 error status and a Www-
Authenticate header, which specifies the authentication protocol the server wants to use.
For example, the server response can contain the following information:
1249985

HTTP/1.0 401 Authorization Required
Www-Authenticate: Basic realm="My Pages"

The realm is meant to be an authentication domain. In practice, it is used in the string that
gets displayed to the user as part of the password prompt. For example, a Web browser will
display this prompt:
1249985

Enter the password for My Pages at www.beedub.com

After getting the user name and password from the user, the Web browser tries its HTTP
request again. This time it includes an Authorization header that contains the user name
and password encoded with base64 encoding. There is no encryption at all — anyone can
decode the string, which is why this is not a strong form of protection. The Standard Tcl
Library includes a base64 package that has base64::encode and base64::decode
procedures. Example 17-13 illustrates the Basic Authentication protocol. It uses the -
headers option to http::geturl that lets you pass additional headers in the request.
1249985

Example 17-13. Basic Authentication using http::geturl

package require base64
package require http
proc BasicAuthentication {url promptProc} {
   set token [http::geturl $url]
   http::wait $token
   if {[string match *401* [http::code $token]]} {
      upvar #0 $token data

      # Extract the realm from the Www-Authenticate line

      array set reply $data(meta)
      if {[regexp {realm=(.*)} $reply(Www-Authenticate) \
             x realm]} {

         # Call back to prompt for username, password

         set answer [$promptProc $realm]
         http::cleanup $token

         # Encode username:password and pass this in
         # the Authorization header

         set auth [base64::encode \
            [lindex $answer 0]:[lindex $answer 1]]
         set token [http::geturl $url -headers \
            [list Authorization "Basic $auth"]]
         http::wait $token
      }
   }
   return $token
}

Example 17-13 takes a promptProc argument that is the name of a procedure to call to get
the username and password. This procedure could display a Tk dialog box, or prompt for
user input from the terminal. In practice, you probably already know the username and
password. In this case, you can skip the initial challenge–response steps and simply supply
the Authorization header on the first request:
1249985

http::geturl $url -headers \
   [list Authorization \
      "Basic [base64::encode $username:$password]"]

