
Chapter 18. TclHttpd Web Server
72

1249985
This chapter describes TclHttpd, a Web server built entirely in Tcl. The Web server can be
used as a standalone server, or it can be embedded into applications to Web-enable them.
TclHttpd provides a Tcl+HTML template facility that is useful for maintaining site-wide look
and feel, and an Application Direct URL that invokes a Tcl procedure in an application.
1249985
TclHttpd started out as about 175 lines of Tcl that could serve up HTML pages and images.
The Tcl socket and I/O commands make this easy, and the C language implementation of
the Tcl runtime library makes the server surprisingly fast. Of course, there are lots of features
in Web servers like Apache or Netscape that were not present in the first prototype. Steve
Uhler took my prototype, refined the HTTP handling, and aimed to keep the basic server
under 250 lines. I went the other direction, setting up a modular architecture, adding in
features found in other Web servers, and adding some interesting ways to connect TclHttpd
to Tcl applications.
1249985
Today TclHttpd is used both as a general-purpose Web server, and as a framework for building
server applications. It implements www.tcl.tk and a number of other general purpose Web
sites. It is also built into several commercial applications such as license servers and mail spam
filters. The server is freely available, just like Tcl itself, and you can use it in any application
without restriction or license fees. Instructions for setting up the TclHttpd on your platform
are given toward the end of the chapter, on page 284. It works on Unix, Windows, and
Macintosh. Using TclHttpd, you can have your own Web server up and running quickly.
1249985
This chapter provides an overview of the server and several examples of how you can use it.
The chapter is not an exhaustive reference to every feature. Instead, it concentrates on a very
useful subset of server features that I use the most.
1249985

Integrating TclHttpd with Your Application
621961

1249985
The bulk of this chapter describes the various ways you can extend the server and integrate
it into your application. TclHttpd is interesting because, as a Tcl script, it is easy to add to your
application. Suddenly your application has an interface that is accessible to Web browsers
in your company's intranet or the global Internet. The Web server provides several ways you
can connect it to your application:
1249985

• Static pages — As a “normal” Web server, you can serve static documents that describe
your application.

• Domain handlers — You can arrange for all URL requests in a section of your Web site to
be handled by your application. This is a very general interface where you interpret what
the URL means and what sort of pages to return to each request. For example, http://
www.tcl.tk/resource is implemented this way. The URL past /resource selects an index
in a simple database, and the server returns a page describing the pages under that
index.

• Application Direct URLs — This is a domain handler that maps URLs onto Tcl procedures.
The form query data that is part of the HTTP GET or POST request is automatically
mapped onto the parameters of the Application Direct procedure. The procedure simply
computes the page as its return value. This is an elegant and efficient alternative to the
CGI interface. For example, in TclHttpd, the URLs under /status report various statistics
about the Web server's operation.

• Document handlers — You can define a Tcl procedure that handles all files of a particular
type. For example, the server has a handler for CGI scripts, HTML files, image maps, and
HTML+Tcl template files.

• HTML+Tcl Templates — These are Web pages that mix Tcl and HTML markup. The server
replaces the Tcl using the subst command and returns the result. The server can cache
the result in a regular HTML file to avoid the overhead of template processing on future
requests. Templates are a great way to maintain the common look and feel to a family
of Web pages, as well as to implement more advanced dynamic HTML features like self-
checking forms.

TclHttpd Architecture
628024
You may find it helpful to read the code to learn more about the features of the server. In this
section, there are references to Tcl files in the source, which are in the lib directory of the
distribution that is on the CD-ROM.
1249985
Figure 18-1 shows the basic components of the server. At the core is the Httpd module
(httpd.tcl), which implements the server side of the HTTP protocol. The “d” in Httpd
stands for daemon, which is the name given to system servers on UNIX. This module manages
network requests, dispatches them to the Url module, and provides routines used to return
the results to requests.
1249985

Figure 18-1. The dotted box represents one application that embeds TclHttpd. Document templates and Application Direct URLs
provide direct connections from an HTTP request to your application. You can also implement completely custom URL handlers.

The Url module (url.tcl) divides the Web site into domains, which are subtrees of the
URL hierarchy provided by the server. The idea is that different domains may have completely
different implementations. For example, the Document domain (doc.tcl) maps its URLs
into files and directories on your hard disk, while the Application Direct domain
(direct.tcl) maps URLs into Tcl procedure calls within your application. The CGI domain
(cgi.tcl) maps URLs onto other programs that compute Web pages.
1249985

Adding Code to TclHttpd
628024
The TclHttpd distribution, which is described in more detail starting at page 284, is set up so
you can easily add code for your application into the server. For simple applications, you
simply put your files into a special directory for custom code, and the server loads them
automatically upon startup. These files should define Tcl procedures and register them as
Domain Handlers, Direct URL handlers, or Document handlers. Example 18-1 implements /
hello/world:
1249985

Example 18-1. The hello.tcl file implements /hello/world

Direct_Url /hello Hello
proc Hello/world {} {
   return "<b>Hello, World!</b>"
}

Suppose you put that file into the directory /tmp/tclhttpd_test. Then you can start the
server like this:
1249985

tclsh8.3 bin/httpd.tcl -library /tmp/tclhttpd_test -debug 1

Now access this URL:
1249985

http://localhost:8015/hello/world

Custom Main Programs
628024
The TclHttpd main program, bin/httpd.tcl, may conflict with the main program of your
existing application. For those applications that embed Tcl interpreters in a more custom
manner, you will need to modify bin/httpd.tcl for use with your application. That script
is not very big, and it is well-commented. The key elements are the Httpd_Server call that
opens the listening socket for the Web server, and the vwait at the very end that activates
the event loop. The rest is all about argument parsing and initializing the various modules
that support the server. It is those aspects that may differ for your custom server application.
1249985

Domain Handlers
621961

1249985
You can implement new kinds of domains that provide your own interpretation of a URL.
This is the most flexible interface available to extend the Web server. You provide a callback
that is invoked to handle every request in a domain, or subtree, of the URL hierarchy. The
callback interprets the URL, computes the page content, and returns the data using routines
from the Httpd module.
1249985
Example 18-2 defines a simple domain that always returns the same page to every request.
The domain is registered with the Url_PrefixInstall command. The arguments to
Url_PrefixInstall are the URL prefix and a callback that is called to handle all URLs that
match that prefix. In the example, all URLs that have the prefix /simple are dispatched to
the SimpleDomain procedure.
1249985
The SimpleDomain handler illustrates several properties of domain handlers. The sock
and suffix arguments to SimpleDomain are appended by Url_Dispatch when it
invokes the domain handler. The sock is the socket connection to the client. The suffix
parameter is the part of the URL after the prefix. For example, if the server receives a request
for the URL /simple/page, then the prefix is /simple and the suffix is /page.
1249985
The prefix argument is defined when the callback is registered with
Url_PrefixInstall. You can specify whatever information you need to pass to the
domain handler. In this simple example, we probably don't need the prefix, but if you
implement several different URL domains with the same handler, then you can pass in the
prefix to distinguish them.
1249985

Example 18-2. A simple URL domain

Url_PrefixInstall /simple [list SimpleDomain /simple]

proc SimpleDomain {prefix sock suffix} {
   upvar #0 Httpd$sock data

   # Generate page header

   set html "<title>A simple page</title>\n"
   append html "<h1>$prefix$suffix</h1>\n"
   append html "<h1>Date and Time</h1>\n"
   append html [clock format [clock seconds]]

   # Display connection state

   append html "<h1>Connection State</h1>"
   append html [html::tableFromArray data border=1]

   # Display query data

   if {[info exist data(query)]} {
      append html "<h1>Query Data</h1>\n"
      append html [html::tableFromList [ncgi::nvlist] border=1]
   }

   Httpd_ReturnData $sock text/html $html
}

Connection State and Query Data
628024
The sock parameter is a handle on the socket connection to the remote client. This variable
is also used to name a state variable that the Httpd module maintains about the connection.
The name of the state array is Httpd$sock. In some cases, you may need access to this
information, and the standard idiom is to use upvar to get a more convenient name for this
array (i.e., data):
1249985

upvar #0 Httpd$sock data

The html and ncgi Packages
628024
The html package provides many procedures useful for generating fragments of HTML. The
html::tableFromArray procedure is used to dump out the connection state in the
data array. Its cousin, html::tableFromList, is used to dump out the query data. The
query data is obtained with the ncgi::nvlist procedure. TclHttpd initializes the ncgi
module so you can use ncgi::nvlist, ncgi::value, and other procedures to access
query data in your domain handlers. Note: it is not necessary to call ncgi::parse as you
would from a CGI script. The html package has some other features, which are described
later, that are very useful when generating HTML forms. These packages are part of the
Standard Tcl Library, tcllib, which can be found along with Tcl and TclHttpd.
1249985

Returning Results
628024
Finally, once the page has been computed, the Httpd_ReturnData procedure is used to
return the page to the client. This takes care of the HTTP protocol as well as returning the
data. There are three related procedures, Httpd_ReturnFile, Httpd_Error, and
Httpd_Redirect. These are summarized in Table 18-1 on page 277.
1249985

Application Direct URLs
621961

1249985
The Application Direct domain implementation provides the simplest way to extend the Web
server. It hides the details associated with query data, decoding URL paths, and returning
results. All you do is define Tcl procedures that correspond to URLs. Their arguments are
automatically matched up to the query data, as shown in Example 13-3 on page 189. The Tcl
procedures compute a string that is the result data, which is usually HTML. That's all there is
to it.
1249985
The name of the Tcl procedure that implements an Application Direct URL is related to the
name of the URL. This way, TclHttpd can automatically look up the Tcl procedure that should
implement a given URL. The Tcl procedure name and the URL have distinct prefixes, but the
suffix is the same. For example, if the Tcl procedure prefix is Demo and the URL prefix is /
demo, then the Demo/time Tcl procedure implements the /demo/time URL. The
Direct_Url procedure sets up the correspondence between the procedures and URLs.
This is shown in Example 18-3:
1249985

Example 18-3. Application Direct URLs

Direct_Url /demo Demo
proc Demo {} {
   return "<html><head><title>Demo page</title></head>\n\
      <body><h1>Demo page</h1>\n\
      <a href=/demo/time>What time is it?</a>\n\
      <form action=/demo/echo>\n\
      Data: <input type=text name=data>\n\
      <br>\n\
      <input type=submit name=echo value='Echo Data'>\n\
      </form>\n\
      </body></html>"
}
proc Demo/time {{format "%H:%M:%S"}} {
   return [clock format [clock seconds] -format $format]
}
proc Demo/echo {args} {
   # Compute a page that echoes the query data
   set html "<head><title>Echo</title></head>\n"
   append html "<body>"
   append html [html::tableFromList $args "border=1"]
   return $html
}

Example 18-3 defines /demo as an Application Direct URL domain that is implemented by
procedures that begin with Demo. There are just three URLs defined:
1249985

/demo
/demo/time
/demo/echo

The /demo page displays a hypertext link to the /demo/time page and a simple form that
will be handled by the /demo/echo page. This page is static, so there is just one return
command in the procedure body. Each line of the string ends with:
1249985

\n\

This is just a formatting trick to let me indent each line in the procedure, without having the
line indented in the resulting string. Actually, the \-newline will be replaced by one space, so
each line will be indented one space. You can leave those off and the page will display the
same in the browser, but when you view the page source, you'll see the indenting. Or you
could not indent the lines in the string, but then your code looks somewhat odd.
1249985
The /demo/time procedure just returns the result of clock format. It doesn't even
bother adding <html>, <head>, or <body> tags, which you can get away with in today's
browsers. A simple result like this is also useful if you are using programs to fetch information
via HTTP requests to your application.
1249985

Using Query Data
628024
Application Direct URL handlers have their parameters automatically assigned to values from
the query data. Like any Tcl procedure, your Application Direct URL procedure can have
named parameters, named parameters with default values, and the args parameter. The
server matches the names of form values with names of your procedure parameters in order
to assign their values. There are three cases:
1249985

• The name of the procedure parameter matches the name of a query data item. The query value is assigned to the parameter.

• The name of the procedure parameter does not appear in the query data. The parameter is assigned the empty string or its
default value, if it has one. The /demo/time procedure is defined with an optional format argument. If a format value is
present in the query data, then it overrides the default value given in the procedure definition.

• The query data item does not match any of the parameters. If the procedure has an args parameter as its last parameter, then
the name and value of the query data item are appended to the args value. Otherwise, the query value is simply ignored. For
example, the /demo/echo procedure's args parameter gets filled in with a name-value list of all query data.

You can see that missing arguments or extra arguments do not cause errors. If you want to
do strict parameter checking, then just use args and check the name-value query list
yourself.
1249985
Here is another example to illustrate the different ways that form data is assigned to
procedure parameters. Suppose you have an Application Direct procedure declared like this:

1249985

proc Demo/param { a b {c cdef} args} { 
body
 }

You could create an HTML form that had elements named a, b, and c, and specified /demo/
param for the ACTION parameter of the FORM tag. Or you could type the following into your
browser to embed the query data right into the URL:
1249985

/demo/param?a=5&b=7&c=red&d=%7ewelch&e=two+words

The ? separates the query data from the URL, and each query item is separated by &. In this
case, when your procedure is called, a is 5, b is 7, c is red, and the args parameter becomes
a list of:
1249985

d ~welch e {two words}

The %7e and the + are special codes for nonalphanumeric characters in the query data. The
+ becomes a space, and the %xx sequence is replaced by the character with character code
xx (e.g., %7e becomes ~). Normally, this encoding is taken care of automatically by the Web
browser when it gets data from a form and passes it to the Web server. However, if you type
query data directly or format URLs with complex query data in them, then you need to encode
special values as we did here. Use the Url_Encode procedure to encode URLs that you put
into Web pages. The Web server automatically decodes the values as it makes the
assignments to the Application Direct URL procedure parameters.
1249985
If a parameter does not match the query data, it gets its default value from the procedure
definition, or it gets the empty string. Consider this example:
1249985

/demo/param?b=5

In this case, a is "", b is 5, c is cdef, and args is an empty list.
1249985

Returning Other Content Types
628024
The default content type for Application Direct URLs is text/html. You can specify other
content types by using a global variable with the same name as your procedure. (Yes, this is
a crude way to craft an interface.) Example 18-4 shows part of the faces.tcl file that
implements an interface to a database of picons — personal icons — that is organized by
user and domain names. The idea is that the database contains images corresponding to
your email correspondents. The Faces_ByEmail procedure, which is not shown, looks up

an appropriate image file. The Application Direct procedure is Faces/byemail, and it sets
the global variable Faces/byemail to the correct Content-Type value based on the
filename extension. The mapping from extension to content type is implemented by the
Mtype procedure (mtype.tcl). MIME is the multimedia content standard for email, and it
originated the various content types now also used in HTTP, hence the term “MIME type.”
1249985

Example 18-4. Alternate types for Application Direct URLs

Direct_Url /faces Faces
proc Faces/byemail {email} {
   global Faces/byemail
   set filename [Faces_ByEmail $email]
   set Faces/byemail [Mtype $filename]
   set in [open $filename]
   fconfigure $in -translation binary
   set X [read $in]
   close $in
   return $X
}

Document Types
621961
The Document domain (doc.tcl) maps URLs onto files and directories. It provides more
ways to extend the server by registering different document type handlers. You can make
up new types to support your application. Example 18-5 shows the pieces needed to create
a handler for a fictitious document type application/myjunk that is invoked to handle
files with the .junk suffix. Use the Mtype_Add procedure to register the mapping from file
suffix to document type:
1249985

Example 18-5. A sample document type handler

# Register the mapping from suffix to MIME type
Mtype_Add application/myjunk .junk

# Define the document handler procedure
#  path is the name of the file on disk
#  suffix is part of the URL after the domain prefix
#  sock is the handle on the client connection

proc Doc_application/myjunk {path suffix sock} {
   upvar #0 Httpd$sock data
   # data(url) is more useful than the suffix parameter.

   # Use the contents of file $path to compute a page
   set contents [
somefunc
 $path]

   # Determine your content type

   set type text/html

   # Return the page
   Httpd_ReturnData $sock $type $data
}

The server finds the document handler in a two-step process. First, the type of a file is
determined by its suffix. The mime.types file contains a map from suffixes to MIME types
such as text/html or image/gif. This map is controlled by the Mtype module in
mtype.tcl. Second, the server checks for a Tcl procedure with the appropriate name:
1249985

Doc_
mimetype

            

The matching procedure, if any, is called to handle the URL request. The procedure should
use routines in the Httpd module to return data for the request. If there is no matching
Doc_mimetype procedure, then the default document handler uses
Httpd_ReturnFile and specifies the Content Type based on the file extension. This is the
heart of the default document handler:
1249985

Httpd_ReturnFile $sock [Mtype $path] $path

As another example, the HTML+Tcl templates use the .tml suffix that is mapped to the
application/x-tcl-template type. You can find the document handler
Doc_application/x-tcl-template in doc.tcl. The TclHttpd distribution also
includes support for files with a .snmp extension that implements a template-based Web
interface to the Scotty SNMP Tcl extension.
1249985

HTML + Tcl Templates
621961

1249985
The template system uses HTML pages that embed Tcl commands and Tcl variable
references. The server replaces these using the subst command and returns the results. The
server comes with a general template system, but using subst is so easy you could create
your own template system. The TclHttpd template framework has these components:
1249985

• Each page.html can have a corresponding page.tml template file. This feature is enabled with the
Doc_CheckTemplates command in the server's configuration file. Normally, the server returns the page.html file unless

the corresponding page.tml file has been modified more recently. In this case, the server processes the template with
subst, caches the result in the page.html file, and returns the result.

• A dynamic template (e.g., a form handler) must be processed each time it is requested. If you put the Doc_Dynamic command
into your page, it turns off the caching of the result in the page.html page. The server responds to a request for a
page.html page by processing the page.tml page. Or you can just reference the page.tml file directly and the server will
always processes the template.

• The server creates a page global Tcl variable that has context about the page being processed. Table 18-6 lists the elements
of the page array.

• The server initializes the env global Tcl variable with similar information, but in the standard way for CGI scripts. Table 18-7
lists the elements of the env array that are set by Cgi_SetEnv in cgi.tcl.

• The server initializes the ncgi module so you can use the ncgi procedures to access query data.

• The server supports per-directory .tml files that contain Tcl source code. These files are designed to contain procedure
definitions and variable settings that are shared among pages. The name of the file is simply “.tml”, with nothing before the
period. This is a standard way to hide files in UNIX, but it can be confusing to talk about the per-directory .tml files and the
page.tml templates that correspond to page.html pages. Before processing each page.tml file, the server will source
the .tml files in all directories leading down to the directory containing the template file. The server compares the modify
time of these files against the template file and will process the template if these .tml files are newer than the cached
page.html file. So, by modifying the .tml file in the root of your URL hierarchy, you invalidate all the cached page.html
files.

Where to Put Your Tcl Code
628024
There are three places you can put the code of your application: directly in your template
pages, in the per-directory .tml files, or in the library directory. There are pros and cons to
each:
1249985

• The library directory is where you should put most of your code. The library directory is specified with the -library command
line argument, and the server loads all files in the library upon startup. The advantage of putting procedure definitions in the
library is that they are defined one time but executed many times. This works well with the Tcl byte-code compiler. The
disadvantage is that if you modify procedures in these files, you have to explicitly source them into the server for these changes
to take effect. You can restart the server, or you can use the /debug/source URL described on page 282 to reload source
files into the running server.

• The .tml files are best for variable definitions that you want to share among pages in a directory, or as a staging area for
procedures during development. The advantage of putting code into the per-directory .tml files is that changes are picked
up immediately with no effort on your part. The server automatically checks if these files are modified and sources them each
time it processes your templates. However, using .tml files tends to scatter your code around the URL tree and can make it
harder to maintain.

• I try to put as little code as possible directly in my page.tml template files. It is awkward to put lots of code there, and you
cannot share procedures and variable definitions easily with other pages. Instead, my goal is to have only procedure calls in
the template files, and put the procedure definitions elsewhere. If you want control structures in your page, such as if and
foreach, you may want to use the version of those commands provided by the html package, as described on page 277.

Templates for Site Structure
628024
The next few examples show a simple template system used to maintain a common “look
and feel” across the pages of a site. The key to a successful template system is a data structure
that defines the structure of the site, and some procedures that generate standard

navigational HTML structure for your pages. Once you do this, then you can easily add new
pages by updating your data structure. The template procedures automatically reformat
your site to include the new pages. Example 18-6 shows a simple one-level site definition
that is kept in the root .tml file. This structure lists the title and URL of each page in the site:
1249985

Example 18-6. A one-level site structure

set site(pages) {
   Home                /index.html
   "Ordering Computers"/ordering.html
   "New Machine Setup" /setup.html
   "Adding a New User" /newuser.html
   "Network Addresses" /network.html
}

Of course, your Web site is likely to have more pages and a more elaborate structure. For
example, you might have several main sections, each with a collection of pages, or even a
three-level hierarchy of pages. Example 18-7 shows another simple data structure to define
a two-level structure. The site(sections) variable stores the names and URLs of the main
sections. For each section, there is an element of site that lists the pages in that section.
Only the About section is shown in the example:
1249985

Example 18-7. A two-level site structure

set site(sections) {
   About     /about
   Products  /products
   Support   /support
}
set site(About) {
   Company   company.html
   Contacts  contacts.html
   Directions   directions.html
}

In practice, you may want to include more information in your data structure to help you
generate HTML. For example, if you have graphics for the main sections, you may need to
record their size. Whatever you need, collect it into your data structures and then generate
the HTML from procedures. You can quickly give your whole site a face lift with new graphics
by changing the template procedures that generate your pages. In contrast, if you hand-
code all your pages, it can take months instead of days.
1249985
Example 18-8 shows a sample template file for the one-level structure shown in Example
18-6. Each page includes two commands, SitePage and SiteFooter, that generate HTML
for the navigational part of the page. Between these commands is regular HTML for the page
content:
1249985

Example 18-8. A HTML + Tcl template file

[SitePage "New Machine Setup"]
This page describes the steps to take when setting up a new
computer in our environment. See
[SiteLink "Ordering Computers"]
for instructions on ordering machines.
<ol>
<li>Unpack and setup the machine.
<li>Use the Network control panel to set the IP address
and hostname.
<!-- Several steps omitted -->
<li>Reboot for the last time.
</ol>
[SiteFooter]

The SitePage procedure takes the page title as an argument. It generates HTML to
implement a standard navigational structure. Example 18-9 has a simple implementation of
SitePage:
1249985

Example 18-9. SitePage template procedure, version 1

proc SitePage {title} {
   global site
   set html "<html><head><title>$title</title></head>\n"
   append html "<body bgcolor=white text=black>\n"
   append html "<h1>$title</h1>\n"
   set sep ""
   foreach {label url} $site(pages) {
      append html $sep
      if {[string compare $label $title] == 0} {
         append html "$label"
      } else {
         append html "<a href='$url'>$label</a>"
      }
      set sep " | "
   }
   return $html
}

The foreach loop that computes the simple menu of links turns out to be useful in many
places. Example 18-10 splits out the loop and uses it in a new version of SitePage along
with the SiteFooter procedure. This version of the templates creates a left column for the
navigation and a right column for the page content. The example also puts a few more visual
elements (e.g., page background color) into the site array so you can easily maintain them:
1249985

Example 18-10. SiteMenu and SiteFooter template procedures

array set site {
   bg        white
   fg        black
   mainlogo  /images/mainLogo.gif
}
proc SitePage {title} {
   global site

   set html "<html><head><title>$title</title></head>\n\
      <body bgcolor=$site(bg) text=$site(fg)>\n\
      <!-- Two Column Layout -->\n\
      <table cellpadding=0>\n\
      <tr><td>\n\
      <!-- Left Column -->\n\
      <img src='$site(mainlogo)'>\n\
      <font size=+1>\n\
      [SiteMenu <br> $site(pages)]\n\
      </font>\n\
      </td><td>\n\
      <!-- Right Column -->\n\
      <h1>$title</h1>\n\
      <p>\n"
   return $html
}
proc SiteFooter {} {
   global site
   set html "<p><hr>\n\
      <font size=-1>[SiteMenu | $site(pages)]</font>\n\
      <!-- Close Right Column -->\n\
      </td></tr></table>\n"
   return $html
}
proc SiteMenu {sep list} {
   global page
   set s ""
   set html ""
   foreach {label url} $list {
      if {[string compare $page(url) $url] == 0} {
         append html $s$label
      } else {
         append html "$s<a href='$url'>$label</a>"
      }
      set s $sep
   }
   return $html
}

There are many other applications for “macros” that make repetitive HTML coding chores
easy. For example, take the SiteLink procedure call in Example 18-8. Instead of hand-
coding the <A> tag with the link to /ordering.html, the page uses the SiteLink
procedure to format the link with a consistent label for the link. Using the procedure also
means that the page will automatically get updated if you change the URL associated with
the ordering page by modifying site(pages). Example 18-11 shows SiteLink:
1249985

Example 18-11. The SiteLink procedure

proc SiteLink {label} {
   global site
   array set map $site(pages)
   if {[info exist map($label)]} {
      return "<a href='$map($label)'>$label</a>"
   } else {
      return $label
   }
}

Using Variables for Important Site Information
628024
Another useful feature of templates is the ability to embed variable references in your pages.
Instead of hard coding the sales phone number, or the current product version number, or
even the product name, you can put variables into your pages. For example, SiteLink and
SitePage take a parameter that is the page title. Instead of hard coding your page titles,
you could keep all of your page titles in an array, and use array references everywhere. That
puts all the text in one place and makes it easy to change. The array definition would look
something like this:
1249985

array set title {
    Home Home
    Order "Ordering Computers"
    Setup "New Machine Setup"
    AddUser "Adding a New User"
    Network "Network Addresses"
}

And the calls to SitePage or SiteLink could be made like this:
1249985

[SitePage $title(Order)]

The .tml pages are a good place to define the variables because the definitions are shared
by all pages in that directory, and in any subdirectories. Also, the definitions in the per-
directory .tml override any definitions that come from the top-level .tml file at the root of
your URL tree. Changing the definition of the variable in the .tml file immediately updates
all the pages that share it.
1249985
The main drawback to variable references is the clash with $ in pricing. If you put $10 into a
page.tml file, it will raise an error (unless the variable 10 is defined). It turns out that you
want to generate prices from some database anyway, so you should avoid hard coding prices
into your pages anyway. It is much better to put [price T-shirt] or $price(T-
shirt) into your page than $10, although if you must do that, just quote the $ with a
backslash, \$10.
1249985

Form Handlers
621961

1249985

HTML forms and form-handling programs go together. The form is presented to the user on
the client machine. The form handler runs on the server after the user fills out the form and
presses the submit button. The form presents input widgets like radiobuttons, checkbuttons,
selection lists, and text entry fields. Each of these widgets is assigned a name, and each widget
gets a value based on the user's input. The form handler is a program that looks at the names
and values from the form and computes the next page for the user to read.
1249985
CGI is a standard way to hook external programs to Web servers for the purpose of processing
form data. CGI has a special encoding for values so that they can be transported safely. The
encoded data is either read from standard input or taken from the command line. The CGI
program decodes the data, processes it, and writes a new HTML page on its standard output.
Chapter 3 describes writing CGI scripts in Tcl.
1249985
TclHttpd provides alternatives to CGI that are more efficient because they are built right into
the server. This eliminates the overhead that comes from running an external program to
compute the page. Another advantage is that the Web server can maintain state between
client requests in Tcl variables. If you use CGI, you must use some sort of database or file
storage to maintain information between requests.
1249985

Application Direct Handlers
628024
The server comes with several built-in form handlers that you can use with little effort.
The /mail/forminfo URL will package up the query data and mail it to you. You use form
fields to set various mail headers, and the rest of the data is packaged up into a Tcl-readable
mail message. Example 18-12 shows a form that uses this handler. Other built-in handlers
are described starting at page 281.
1249985

Example 18-12. Mail form results with /mail/forminfo

<form action=/mail/forminfo method=post>
   <input type=hidden name=sendto value=mailreader@my.com>
   <input type=hidden name=subject value="Name and Address">
   <table>
      <tr><td>Name</td><td><input name=name></td></tr>
      <tr><td>Address</td><td><input name=addr1></td></tr>
      <tr><td> </td><td><input name=addr2></td></tr>
      <tr><td>City</td><td><input name=city></td></tr>
      <tr><td>State</td><td><input name=state></td></tr>
      <tr><td>Zip/Postal</td><td><input name=zip></td></tr>
      <tr><td>Country</td><td><input name=country></td></tr>
   </table>
</form>

The mail message sent by /mail/forminfo is shown in Example 18-13.
1249985

Example 18-13. Mail message sent by /mail/forminfo

To: mailreader@my.com
Subject: Name and Address

data {
   name   {Joe Visitor}
   addr1  {Acme Company}
   addr2  {100 Main Street}
   city   {Mountain View}
   state  California
   zip    12345
   country   USA
}

The email message is designed to be easily processed by a Tcl program. You can use a mail
processor like procmail to filter all mail with a given Subject or To field to a program for
processing. It is easy to write a script that strips the headers, defines a data procedure, and
uses eval to process the message body. Whenever you send data via email, if you format it
with Tcl list structure, you can process it quite easily. The basic structure of such a mail reader
procedure is shown in Example 18-14:
1249985

Example 18-14. Processing mail sent by /mail/forminfo

# Assume the mail message is on standard input

set X [read stdin]

# Strip off the mail headers, when end with a blank line
if {[regsub {.*?\n\ndata} $X {data} X] != 1} {
   error "Malformed mail message"
}
proc data {fields} {
   foreach {name value} $fields {
      # Do something
   }
}
# Process the message.
eval $X

The raw eval in the mail handler is dangerous. It will be fine if the only source of email to
that program is the /mail/forminfo URL handler. However, an attacker could send you
an email that results in arbitrary Tcl commands being evaluated by your mail processor. The
safe way to process the email is with a safe interpreter, which is described in Chapter 19.
Example 18-15 adds just a few commands to create a safe interpreter for processing the
incoming data. The data command is evaluated in the trusted interpreter by the alias
mechanism. All other commands in the email are evaluated in the safe interpreter, and any
malicious commands simply raise Tcl errors but cause no harm:
1249985

Example 18-15. Processing mail sent by /mail/forminfo, Safe-Tcl version

# Assume the mail message is on standard input

set X [read stdin]

# Strip off the mail headers, when end with a blank line
if {[regsub {.*?\n\ndata} $X {data} X] != 1} {
   error "Malformed mail message"
}
proc data {fields} {
   foreach {name value} $fields {
      # Do something
   }
}
# Create the safe interpreter
set i [interp create -safe]

# Link the data command in the safe interpreter to the
# data procedure in this interpreter
interp alias $i data {} data

# Process the message in the safe interpreter
interp eval $i $X

Template Form Handlers
628024
The drawback of using Application Direct URL form handlers is that you must modify their
Tcl implementation to change the resulting page. Another approach is to use templates for
the result page that embed a command that handles the form data. The Mail_FormInfo
procedure, for example, mails form data. It takes no arguments. Instead, it looks in the query
data for sendto and subject values, and if they are present, it sends the rest of the data
in an email. It returns an HTML comment that flags that mail was sent.
1249985
When you use templates to process form data, you need to turn off result caching because
the server must process the template each time the form is submitted. To turn off caching,
embed the Doc_Dynamic command into your form handler pages, or set the page
(dynamic) variable to 1. Alternatively, you can simply post directly to the file.tml page
instead of to the file.html page.
1249985

Self-Posting Forms
628024
This section illustrates a self-posting form. This is a form on a page that posts the form data
to back to the same page. The page embeds a Tcl command to check its own form data. Once
the data is correct, the page triggers a redirect to the next page in the flow. This is a powerful
technique that I use to create complex page flows using templates. Of course, you need to
save the form data at each step. You can put the data in Tcl variables, use the data to control
your application, or store it into a database. TclHttpd comes with a Session module, which

is one way to manage this information. For details, you should scan the session.tcl file
in the distribution.
1249985
Example 18-16 shows the Form_Simple procedure that generates a simple self-checking
form. Its arguments are a unique ID for the form, a description of the form fields, and the URL
of the next page in the flow. The field description is a list with three elements for each field:
a required flag, a form element name, and a label to display with the form element:
1249985

Example 18-16. A self-checking form procedure

proc Form_Simple {id fields nextpage} {
   global page
   if {![html::varEmpty formid]} {
      # Incoming form values, check them
      set check 1
   } else {
      # First time through the page
      set check 0
   }
   set html "<!-- Self-posting. Next page is $nextpage -->\n"
   append html "<form action=\"$page(url)\" method=post>\n"
   append html "<input type=hidden name=formid value=$id>\n"
   append html "<table border=1>\n"
   foreach {required key label} $fields {
      append html "<tr><td>"
      if {$check && $required && [html::varEmpty $key]} {
         lappend missing $label
         append html "<font color=red>*</font>"
      }
      append html "</td><td>$label</td>\n"
      append html "<td><input [html::formValue $key]></td>\n"
      append html "</tr>\n"
   }
   append html "</table>\n"
   if {$check} {
      if {![info exist missing]} {

         # No missing fields, so advance to the next page.
         # In practice, you must save the existing fields
         # at this point before redirecting to the next page.

         Doc_Redirect $nextpage
      } else {
         set msg "<font color=red>Please fill in "
         append msg [join $missing ", "]
         append msg "</font>"
         set html <p>$msg\n$html
      }
   }
   append html "<input type=submit>\n</form>\n"
   return $html
}

The Form_Simple procedure does two things at once: it computes the HTML form, and it
also checks if the required fields are present. It uses some procedures from the html module
to generate form elements that retain values from the previous page. If all the required fields
are present, then it triggers a redirect by calling Doc_Redirect. Example 18-17 shows a
page template that calls Form_Simple with the required field description:
1249985

Example 18-17. A page with a self-checking form

<html><head>
   <title>Name and Address Form</title>
</head>
<body bgcolor=white text=black>
   <h1>Name and Address</h1>
   Please enter your name and address.
   [Form_Simple nameaddr {
      1 name    "Name"
      1 addr1   "Address"
      0 addr2"  "Address"
      1 city    "City"
      0 state   "State"
      1 zip     "Zip Code"
      0 country "Country"
   } nameok.html]
</body></html>

The html Package
628024
The Standard Tcl Library, tcllib, includes an html package that is designed to support page
generation and self-posting forms. The html package works in conjunction with the ncgi
package, which was introduced in Chapter 3. The Form_Simple procedure uses
html::varEmpty to test if particular form values are present in the query data. For example,
it tests to see whether the formid field is present so that the procedure knows whether or
not to check for the rest of the fields. The html::formValue procedure is useful for
constructing form elements on self-posting form pages. It returns:
1249985

name="
name
" value="
value
"

The value is the value of form element name based on incoming query data, or just the
empty string if the query value for name is undefined. As a result, the form can post to itself
and retain values from the previous version of the page. It is used like this:
1249985

<input type=text [html::formValue 
name
]>

The html::checkValue and html::radioValue procedures are similar to
html::formValue, but are designed for checkbuttons and radio buttons. The
html::select procedure formats a selection list and highlights the selected values.
1249985

The html package includes a versions of foreach and if that are designed for use in
templates. These commands perform a subst on their body instead of evaluating it. This
lets you put HTML with variable and command references into the body to build up results.
Example 18-18 shows the html::foreach procedure used to generate a table with several
rows. Note that you don't have to worry about the $ in the prices because they are inside the
braces of the html::foreach value list:
1249985

Example 18-18. Generating a table with html::foreach

<TABLE BORDER=1>
[html::foreach {product price} {
   T-Shirt   $10.00
   YoYo      $7.50
   Footbag   $15.00
} {
   <TR>
      <TD>$product</TD>
      <TD ALIGN=RIGHT><FONT FACE=courier>$price</FONT></TD>
   </TR>
}
</TABLE>

Programming Reference
621961
This section summarizes many of the more useful functions defined by the server. These
tables are not complete, however. You are encouraged to read through the code to learn
more about the features offered by the server. A simple naming convention is used to
distinguish procedures that are private to a file (e.g., HttpdEvent) and procedures that are
meant to be used by other modules or by the main application (e.g., Httpd_Server). The
underscore after the module prefix indicates that the procedure is public.
1249985
This section does not detail the ncgi and html packages, which are quite useful to the
TclHttpd programmer. There are doc files that come with tcllib, and you can find man pages
for the tcllib packages in the www.tcl.tk manual section.
1249985
Table 18-1 shows Httpd functions used when returning pages to the client.
1249985

Table 18-1. Httpd support procedures

Httpd_Error sock code
1249985

Returns a simple error page to the client. The
code is a numeric error code such as 404 or
500.

1249985

Httpd_ReturnData sock type data
1249985

Returns a page with Content-Type type and
content data.
1249985

Httpd_ReturnFile sock type file
1249985

Returns a file with Content-Type type.
1249985

Httpd_Redirect newurl sock
1249985

Generates a 302 error return with a Location
of newurl.
1249985

Httpd_SelfUrl url
1249985

Expands url to include the proper http://
server:port prefix to reference the
current server.
1249985

Table 18-2 summarizes a few useful procedures provided by the Url module (url.tcl).
The Url_DecodeQuery is used to decode query data into a Tcl-friendly list. The
Url_Encode procedure is useful when encoding values directly into URLs. URL encoding is
discussed in more detail on page 262
1249985

Table 18-2. Url support procedures

Url_DecodeQuery query
1249985

Decodes a www-url-encoded query
string and returns a name, value list.
Depreciated. This is equivalent to
ncgi::nvlist, which takes no
arguments.
1249985

Url_Encode value
1249985

Returns value encoded according to the
www-url-encoded standard.
1249985

Url_PrefxInstall prefix
handler ?-thread bool? ?-callback
cmd??-readpost bool?
1249985

Registers handler as the handler for all
URLs that begin with prefix. The
handler is invoked with two additional
arguments: sock, the handle to the client,
and suffix, the part of the URL after
prefix. Use -thread 1 to have the

handler run in a worker thread. Use -
callback cmd to register a callback
invoked at the very end of URL processing.
Use -readpost 0 to disable pre-
reading post data.
1249985

The Doc module procedures for configuration are listed in Table 18-3.
1249985

Table 18-3. Doc procedures for configuration

Doc_Root ?directory?
1249985

Sets or queries the directory that
corresponds to the root of the URL hierarchy.
1249985

Doc_AddRoot virtual directory 
1249985

Maps the file system directory into the URL
subtree starting at virtual.
1249985

Doc_ErrorPage file
1249985

Specifies a file relative to the document root
used as a simple template for error messages.
This is processed by DocSubstSystem file in
doc.tcl.
1249985

Doc_CheckTemplates how
1249985

If how is 1, then .html files are compared
against corresponding .tml files and
regenerated, if necessary.
1249985

Doc_IndexFile pattern
1249985

Registers a file name pattern that will be
searched for the default index file in directories.
1249985

Doc_NotFoundPage file
1249985

Specifies a file relative to the document root
used as a simple template for page not found
messages. This is processed by
DocSubstSystem file in doc.tcl.
1249985

Doc_PublicHtml dirname
1249985

Defines the directory used for each user's home
directory. When a URL such as ~user is

specified, the dirname under their home
directory is accessed.
1249985

Doc_TemplateLibrary directory
1249985

Adds directory to the auto_path so that
the source files in it are available to the server.
1249985

Doc_TemplateInterp interp
1249985

Specifies an alternate interpreter in which to
process document templates (i.e., .tml files.)
1249985

Doc_Webmaster ?email?
1249985

Sets or queries the email for the Webmaster.
1249985

The Doc module procedures for generating results are listed in Table 18-4
1249985

Table 18-4. Doc procedures for generating responses

Doc_Error sock errorInfo
1249985

Generates a 500 response on sock based on the
template registered with Doc_ErrorPage.
errorInfo is a copy of the Tcl error trace after the
error.
1249985

Doc_NotFound sock
1249985

Generates a 404 response on sock by using the
template registered with Doc_NotFoundPage.
1249985

Doc_Subst sock file ?interp?
1249985

Performs a subst on the file and return the
resulting page on sock. interp specifies an
alternate Tcl interpreter.
1249985

The Doc module also provides procedures for cookies and redirects that are useful in
document templates. These are described in Table 18-5.
1249985

Table 18-5. Doc procedures that support template processing

Doc_Coookie name
1249985

Returns the cookie name passed to the
server for this request, or the empty
string if it is not present.
1249985

Doc_Dynamic
1249985

Turns off caching of the HTML result.
Meant to be called from inside a page
template.
1249985

Doc_IsLinkToSelf url
1249985

Returns 1 if the url is a link to the
current page.
1249985

Doc_Redirect newurl
1249985

Raises a special error that aborts
template processing and triggers a
page redirect to newurl.
1249985

Doc_SetCookie -name name -value
value -path path -domain domain -
expires date
1249985

Sets cookie name with the given
value that will be returned to the
client as part of the response. The
path and domain restrict the scope of
the cooke. The date sets an expiration
date.
1249985

Table 18-6 shows the initial elements of the page array that are defined during the processing
of a template.
1249985

Table 18-6. Elements of the page array

query
1249985

The decoded query data in a name, value list. Also available through ncgi.
1249985

dynamic
1249985

If 1, the results of processing the template are not cached in the
corresponding .html file.
1249985

filename
1249985

The file system pathname of the requested file (e.g., /usr/local/htdocs/
tclhttpd/index.html).
1249985

template
1249985

The file system pathname of the template file (e.g., /usr/local/htdocs/
tclhttpd/index.tml).
1249985

url The part of the URL after the server name (e.g., /tclhttpd/index.tml).

1249985 1249985

root
1249985

A relative path from the template file back to the root of the URL tree.
1249985

Table 18-7 shows the elements of the env array. These are defined during CGI requests,
Application Direct URL handlers, and page template processing:
1249985

Table 18-7. Elements of the env array

AUTH_TYPE
1249985

Authentication protocol (e.g., Basic).
1249985

CONTENT_LENGTH
1249985

The size of the query data.
1249985

CONTENT_TYPE
1249985

The type of the query data.
1249985

DOCUMENT_ROOT
1249985

File system pathname of the document root.
1249985

GATEWAY_INTERFACE
1249985

Protocol version, which is CGI/1.1.
1249985

HTTP_ACCEPT
1249985

The Accept headers from the request.
1249985

HTTP_AUTHORIZATION
1249985

The Authorization challenge from the request.
1249985

HTTP_COOKIE
1249985

The cookie from the request.
1249985

HTTP_FROM
1249985

The From: header of the request.

HTTP_REFERER
1249985

The Referer indicates the previous page.
1249985

HTTP_USER_AGENT
1249985

An ID string for the Web browser.
1249985

PATH_INFO
1249985

Extra path information after the template file.
1249985

PATH_TRANSLATED The extra path information appended to the document root.

Chapter 18. TclHttpd Web Server Page 27 Return to Table of Contents

Chapter 18. TclHttpd Web Server
Practical  Programming in  Tcl  and Tk,  Fourth  Edition By Ken Jones,  Jeffrey Hobbs,  Brent  B.  Welch ISBN:
0-13-038560-3 Publisher: Prentice Hall

Prepared for Ronald Fischer, Safari ID: ronald.fischer@fusshuhn.de

Print Publication Date: 6/10/2003 User number: 628024 Copyright 2007, Safari Books Online, LLC.
Reproduction, transmission and/or redistribution in any form by any means without the prior written permission from the publisher is prohibited.



1249985 1249985

QUERY_STRING
1249985

The form query data.
1249985

REMOTE_ADDR
1249985

The client's IP address.
1249985

REMOTE_USER
1249985

The remote user name specified by Basic authentication.
1249985

REQUEST_METHOD
1249985

GET, POST, or HEAD.
1249985

REQUEST_URI
1249985

The complete URL that was requested.
1249985

SCRIPT_NAME
1249985

The name of the current file relative to the document root.
1249985

SERVER_NAME
1249985

The server name, e.g., www.beedub.com.
1249985

SERVER_PORT
1249985

The server's port, e.g., 80.
1249985

SERVER_PROTOCOL
1249985

The protocol (e.g., http or https).
1249985

SERVER_SOFTWARE
1249985

A software version string for the server.
1249985

Standard Application Direct URLs
621961
The server has several modules that provide Application Direct URLs. These Application
Direct URLs let you control the server or examine its state from any Web browser. You can
look at the implementation of these modules as examples for your own application.
1249985

Status
628024
The /status URL is implemented in the status.tcl file. The status module implements
the display of hit counts, document hits, and document misses (i.e., documents not found).

The Status_Url command enables the Application Direct URLs and assigns the top-level
URL for the status module. The default configuration file contains this command:
1249985

Status_Url /status

Table 18-8 shows the URLs implemented by the status module:
1249985

Table 18-8. Status Application Direct URLs

/status
1249985

Main status page showing summary counters and hit count
histograms.
1249985

/status/doc
1249985

Shows hit counts for each page. This page lets you sort by name or
hit count, and limit files by patterns.
1249985

/status/domain
1249985

Shows hit counts for each domain in the server.
1249985

/status/hello
1249985

A trivial URL that returns "hello".
1249985

/status/notfound
1249985

Shows miss counts for URLs that users tried to fetch.
1249985

/status/size
1249985

Displays an estimated size of Tcl code and Tcl data used by the
TclHttpd program.
1249985

/status/text
1249985

This is a version of the main status page that doesn't use the
graphical histograms of hit counts.
1249985

Debugging
628024
The /debug URL is implemented in the debug.tcl file. The debug module has several
useful URLs that let you examine variable values and other internal state. It is turned on with
this command in the default configuration file:
1249985

Debug_Url /debug

Table 18-9 lists the /debug URLs. These URLs often require parameters that you can specify
directly in the URL. For example, the /debug/echo URL echoes its query parameters:
1249985

http://yourserver:port/debug/echo?name=value&name2=val2

Note: The debug URL is active in the default configuration. If it makes you nervous, then
delete the call to Debug_Url from the httpdthread.tcl file.
The sample URL tree that is included in the distribution includes the file htdocs/
hacks.html. This file has several small forms that use the /debug URLs to examine
variables and source files. It may seem dangerous to have these facilities, but I reason that
because my source directories are under my control, it cannot hurt to reload any source files.
In general, the library scripts contain only procedure definitions and no global code that
might reset state inappropriately. In practice, the ability to tune (i.e., fix bugs) in the running
server has proven useful to me on many occasions. It lets you evolve your application without
restarting it!
1249985

Table 18-9. Debug Application Direct URLs

/debug/after
1249985

Lists the outstanding after events.
1249985

/debug/dbg
1249985

Connects to TclPro Debugger. This takes a host and port
parameter. You need to install prodebug.tcl from TclPro into
the server's script library directory.
1249985

/debug/echo
1249985

Echoes its query parameters. Accepts a title parameter.
1249985

/debug/errorInfo
1249985

Displays the errorInfo variable along with the server's version
number and Webmaster email. Accepts title and errorInfo
arguments.
1249985

/debug/parray
1249985

Displays a global array variable. The name of the variable is
specified with the aname parameter.
1249985

/debug/pvalue
1249985

A more general value display function. The name of the variable is
specified with the aname parameter. This can be a variable name,
an array name, or a pattern that matches several variable names.
1249985

/debug/raise
1249985

Raises an error (to test error handling). Any parameters become the
error string.
1249985

/debug/source
1249985

Sources a file from either the server's main library directory or the
Doc_TemplateLibrary directory. The file is specified with the
source parameter.
1249985

Example 18-19 shows the implementation of /debug/source. You can see that it limits the
files to the main script library and to the script library associated with document templates.
1249985

Example 18-19. The /debug/source Application Direct URL implementation

proc Debug/source {source} {
   global Httpd Config errorInfo
   set source [file tail $source]
   set dirlist $Httpd(library)   ;# TclHttpd implementation
   lappend dirlist $Config(lib)  ;# Application custom code
   foreach dir $dirlist {
      set file [file join $dir $source]
      if {[file exists $file]} break
   }
   set error [catch {uplevel #0 [list source $file]} result]
   set html "<title>Source $source</title>\n"
   if {$error} {
      append html "<H1>Error in $source</H1>\n"
      append html "<pre>$result<p>$errorInfo</pre>"
   } else {
      append html "<H1>Reloaded $source</H1>\n"
      append html "<pre>$result</pre>"
   }
   return $html
}

Sending Email
628024
The /mail URL is implemented in the mail.tcl file. The mail module implements various
form handlers that email form data. Currently, it is UNIX-specific because it uses /usr/lib/
sendmail to send the mail. It is turned on with this command in the default configuration
file:
1249985

Mail_Url /mail

The Application Direct URLs shown in Table 18-10 are useful form handlers. You can specify
them as the ACTION parameter in your <FORM> tags. The mail module provides two Tcl
procedures that are generally useful. The MailInner procedure is the one that sends mail.
It is called like this:

1249985

Table 18-10. Application Direct URLS that email form results

/mail/bugreport
1249985

Sends email with the errorInfo from a server error. It takes an
email parameter for the destination address and an errorInfo
parameter. Any additional arguments get included into the
message.
1249985

/mail/forminfo
1249985

Sends email containing form results. It requires these parameters:
sendto for the destination address, subject for the mail subject,
href and label for a link to display on the results page. Any
additional arguments are formatted with the Tcl list command for
easy processing by programs that read the mail.
1249985

/mail/formdata
1249985

This is an older form of /mail/forminfo that doesn't format the
data into Tcl lists. It requires only the email and subject
parameters. The rest are formatted into the message body.
1249985

MailInner 
sendto subject from type body

            

The sendto and from arguments are email addresses. The type is the MIME type (e.g.,
text/plain or text/html) and appears in a Content-Type header. The body contains
the mail message without any headers.
1249985
The Mail_FormInfo procedure is designed for use in HTML+Tcl template files. It takes no
arguments but instead looks in current query data for its parameters. It expects to find the
same arguments as the /mail/forminfo direct URL. Using a template with
Mail_FormInfo gives you more control over the result page than posting directly to /
mail/forminfo, and is illustrated in Example 18-12 on page 272.
1249985

The TclHttpd Distribution
621961
Get the TclHttpd distribution from the CD-ROM, or find it on the Internet at:
1249985

ftp://ftp.tcl.tk/pub/tcl/httpd/

http://www.tcl.tk/software/tclhttpd/
http://www.sourceforge.net/projects/tclhttpd

Quick Start
628024
Unpack the tar file or the zip file, and you can run the server from the httpd.tcl script in
the bin directory. On UNIX:
1249985

tclsh bin/httpd.tcl -port 80

This command will start the Web server on the standard port (80). On UNIX, you need to be
root to run a server on this port. By default TclHttpd uses port 8015 instead. If you run it with
the -help flag, it will tell you what command line options are available. If you use wish instead
of tclsh, then a simple Tk user interface is displayed that shows how many hits the server is
getting.
1249985
On Windows, you can double-click the httpd.tcl script to start the server. It will use
wish and display the user interface. Again it will start on port 8015. You will need to create a
shortcut that passes the -port argument, or edit the associated configuration file to change
this. Configuring the server is described later.
1249985
Once you have the server running, you can connect to it from your Web browser. Use this
URL if you are running on the default (nonstandard) port:
1249985

http://hostname:8015/

If you are running without a network connection, you may need to specify 127.0.0.1 for
the hostname. This is the "localhost" address and will bypass the network subsystem.
1249985

http://127.0.0.1:8015/

Inside the Distribution
628024
The TclHttpd distribution is organized into the following directories:
1249985

• bin — This has sample start-up scripts and configuration files. The httpd.tcl script
runs the server. The tclhttpd.rc file is the standard configuration file.

• bin/mini — This has a few tiny versions of the server that provide a basic server in
about 300 lines of code. Use these as a starting point by modifying the
HttpdRespond procedure.

• bin/test — This has a number of test scripts, including the torture.tcl file that
can fetch many URLs at once from a server.

• certs — This has sample certificates you can use to test a secure server for https URLs.
If you have your own server certificates, put the server.pem file here.

• config — This contains autoconf support used by C extensions you can build with the
server.

• custom — This is where you put your own custom code. Files here are automatically
loaded by the server on startup. This contains a few samples.

• doc — This has a UNIX-style manual page for how to run the server.
• htaccess — This has sample access control files.
• htdocs — This is a sample URL tree that demonstrates the features of the Web server.

There is also some documentation there. One directory to note is htdocs/libtml,
which is the standard place to put site-specific Tcl scripts used with the Tcl+HTML
template facility.

• lib — This has all the Tcl sources. In general, each file provides a package. You will see
the package require commands partly in bin/httpd.tcl and partly in bin/
httpdthread.tcl.

• src — There are a few C source files for a some optional packages. These have been
precompiled for some platforms, and you can find the compiled libraries under src/
Solaris and src/Linux.

Server Configuration
621961
TclHttpd configures itself with two main steps: setting configuration parameters and loading
packages. The configuration step uses a configuration file and command line arguments to
set basic configuration parameters. The default configuration file is named tclhttpd.rc
in the same directory as the start-up script (i.e., bin/tclhttpd.rc). Specify an alternate
configuration file with the -config command line argument. You can override the
configuration file with additional command line arguments, which are described in Table
18-11. The configuration values from the file and the command line are copied into the
Config Tcl array.
1249985
Package loading is split into two parts. The main bin/httpd.tcl script loads some core
packages. The rest are loaded in the bin/httpdthread.tcl script. The reason for the split
is to try to isolate the core of the server from application-specific functions. In addition, in
the threaded version of the server, every thread loads and runs the bin/

httpdthread.tcl script. You can specify an alternate package loading script with the -
main command line argument.
1249985
For example, to start the server for the document tree under /usr/local/htdocs and
your own email address as Webmaster, you can execute this command to start the server:
1249985

tclsh httpd.tcl -docRoot /usr/local/htdocs -webmaster welch

If you are using the Tclkit version described in Chapter 22:
1249985

tclkit tclhttpd.kit -docRoot /usr/local/htdocs -webmaster welch

Alternatively, you can put these settings into a configuration file, and start the server with
that configuration file:
1249985

tclsh httpd.tcl -config mytclhttpd.rc

Command Line Arguments
628024
There are several parameters you may need to set for a standard Web server. These are shown
below in Table 18-11. The command line values are mapped into the Config array by the
httpd.tcl startup script.
1249985

Table 18-11. Basic TclHttpd parameters

Parameter
1249985

Command Option
1249985

Config Variable
1249985

Port number. The default is 8015.
1249985

-port number
1249985

Config(port)
1249985

Server name. The default is [info
hostname].
1249985

-name name
1249985

Config(name)
1249985

IP address. The default is 0, for "any
address".
1249985

-ipaddr address
1249985

Config(ipaddr)
1249985

Directory of the root of the URL tree.
The default is the htdocs directory.

-docRoot directory
1249985

Config(docRoot)
1249985

Parameter
1249985

Command Option
1249985

Config Variable
1249985

1249985

User ID of the TclHttpd process. The
default is 50. (UNIX only.)
1249985

-uid uid
1249985

Config(uid)
1249985

Group ID of the TclHttpd process. The
default is 100. (UNIX only.)
1249985

-gid gid
1249985

Config(gid)
1249985

Webmaster email. The default is
webmaster.
1249985

-webmaster email
1249985

Config(webmaster)
1249985

Configuration file. The default is
tclhttpd.rc.
1249985

-config filename
1249985

Config(file)
1249985

Directory containing custom code.
The server loads all files found in this
directory.
1249985

-library directory
1249985

Config(library)
1249985

Server Name and Port
628024
The name and port parameters define how your server is known to Web browsers. The URLs
that access your server begin with:
1249985

http://name:port/

If the port number is 80, you can leave out the port specification. The call that starts the server
using these parameters is found in httpd.tcl as:
1249985

Httpd_Server $Config(name) $Config(port) $Config(ipaddr)

Specifying the IP address is necessary only if you have several network interfaces (or several
IP addresses assigned to one network interface) and want the server to listen to requests on
a particular network address. Otherwise, by default, the server accepts requests from any
network interface.
1249985

User and Group ID
628024
The user and group IDs are used on UNIX systems with the setuid and setgid system calls.
This lets you start the server as root, which is necessary to listen on port 80, and then switch
to a less privileged user account. If you use Tcl+HTML templates that cache the results in
HTML files, then you need to pick an account that can write those files. Otherwise, you may
want to pick a very unprivileged account.
1249985
The setuid function is available through the TclX (Extended Tcl) id command, or through
a setuid extension distributed with TclHttpd under the src directory. If either of these
facilities is not available, then the attempt to change user ID gracefully fails. See the
README file in the src directory for instructions on compiling and installing the extensions
found there.
1249985

Webmaster Email
628024
The Webmaster email address is used for automatic error reporting in the case of server errors.
This is defined in the configuration file with the following command:
1249985

Doc_Webmaster $Config(webmaster)

If you call Doc_Webmaster with no arguments, it returns the email address you previously
defined. This is useful when generating pages that contain mailto: URLs with the
Webmaster address.
1249985

Document Root
628024
The document root is the directory that contains the static files, templates, CGI scripts, and
so on that make up your Web site. By default, the httpd.tcl script uses the htdocs directory
next to the directory containing httpd.tcl. It is worth noting the trick used to locate this
directory:
1249985

file join [file dirname [info script]] ../htdocs

The info script command returns the full name of the http.tcl script, file
dirname computes its directory, and file join finds the adjacent directory. The

path ../htdocs works with file join on any platform. The default location of the
configuration file is found in a similar way:
1249985

file join [file dirname [info script]] tclhttpd.rc

The configuration file initializes the document root with this call:
1249985

Doc_Root $Config(docRoot)

If you need to find out what the document root is, you can call Doc_Root with no arguments
and it returns the directory of the document root. If you want to add additional document
trees into your Web site, you can do that with a call like this in your configuration file:
1249985

Doc_AddRoot 
directory urlprefix

            

Other Document Settings
628024
The Doc_IndexFile command sets a pattern used to find the index file in a directory. The
command used in the default configuration file is:
1249985

Doc_IndexFile index.{htm,html,tml,subst}

If you invent other file types with different file suffixes, you can alter this pattern to include
them. This pattern will be used by the Tcl glob command.
1249985
The Doc_PublicHtml command is used to define "home directories" on your HTML site. If
the URL begins with ~username, then the Web server will look under the home directory
of the user for a particular directory. The command in the default configuration file is:
1249985

Doc_PublicHtml public_html

For example, if my home directory is /home/welch, then the URL ~welch maps to the
directory /home/welch/public_html. If there is no Doc_PublicHtml command, then
this mapping does not occur.
1249985
You can register two special pages that are used when the server encounters an error and
when a user specifies an unknown URL. The default configuration file has these commands:
1249985

Doc_ErrorPage error.html
Doc_NotFoundPage notfound.html

These files are treated like templates in that they are passed through subst in order to
include the error information or the URL of the missing page. These are pretty crude
templates compared to the templates described earlier. You can count only on the Doc and
Httpd arrays being defined. Look at the Doc_SubstSystemFile in doc.tcl for the truth
about how these files are processed.
1249985

Document Templates
628024
The template mechanism has two main configuration options. The first specifies an
additional library directory that contains your application-specific scripts. This lets you keep
your application-specific files separate from the TclHttpd implementation. The command in
the default configuration file specifies the libtml directory of the document tree:
1249985

Doc_TemplateLibrary [file join $Config(docRoot) libtml]

You can also specify an alternate Tcl interpreter in which to process the templates. The default
is to use the main interpreter, which is named {} according to the conventions described in
Chapter 19.
1249985

Doc_TemplateInterp {}

Log Files
628024
The server keeps standard format log files. The Log_SetFile command defines the base
name of the log file. The default configuration file uses this command:
1249985

Log_SetFile /tmp/log$Config(port)_

By default, the server rotates the log file each night at midnight. Each day's log file is suffixed
with the current date (e.g., /tmp/logport_990218.) The error log, however, is not rotated,
and all errors are accumulated in /tmp/logport_error.
1249985
The log records are normally flushed every few minutes to eliminate an extra I/O operation
on each HTTP transaction. You can set this period with Log_FlushMinutes. If minutes is
0, the log is flushed on every HTTP transaction. The default configuration file contains:
1249985

Log_FlushMinutes 1

CGI Directories
628024
You can register a directory that contains CGI programs with the Cgi_Directory
command. This command has the interesting effect of forcing all files in the directory to be
executed as CGI scripts, so you cannot put normal HTML files there. The default configuration
file contains:
1249985

Cgi_Directory /cgi-bin

This means that the cgi-bin directory under the document root is a CGI directory. If you
supply another argument to Cgi_Directory, then this is a file system directory that gets
mapped into the URL defined by the first argument. You can also put CGI scripts into other
directories and use the .cgi suffix to indicate that they should be executed as CGI scripts.
1249985
The cgi.tcl file has some additional parameters that you can tune only by setting some
elements of the Cgi Tcl array. See the comments in the beginning of that file for details.
1249985

