
Chapter 3. The Guestbook CGI Application
711

1249985
This chapter presents a simple Tcl program that computes a Web page. The chapter provides a brief
background to HTML and the CGI interface to Web servers. The chapter uses the ncgi package
from the standard Tcl library.
1249985
This chapter presents a complete, but simple, guestbook program that computes an HTML
document, or Web page, based on the contents of a simple database. The basic idea is that a user
with a Web browser visits a page that is computed by the program. The details of how the page
gets from your program to the user with the Web browser vary from system to system. The Tcl
Web Server described in Chapter 18 comes with this guestbook example already set up. You can
also use these scripts on your own Web server, but you will need help from your Webmaster to set
things up.
1249985
The chapter provides a very brief introduction to HTML and CGI programming. HTML is a way
to specify text formatting, including hypertext links to other pages on the World Wide Web. CGI
is a standard for communication between a Web server that delivers documents and a program that
computes documents for the server. There are many books on these subjects alone.
1249985
A guestbook is a place for visitors to sign their name and perhaps provide other information. We
will build a guestbook that takes advantage of the World Wide Web. Our guests can leave their
address as a Universal Resource Location (URL). The guestbook will be presented as a page that
has hypertext links to all these URLs so that other guests can visit them. The program works by
keeping a simple database of the guests, and it generates the guestbook page from the database.
1249985
The Tcl scripts described in this chapter use commands and techniques that are described in more
detail in later chapters. The goal of the examples is to demonstrate the power of Tcl without
explaining every detail. If the examples in this chapter raise questions, you can follow the references
to examples in other chapters that do go into more depth.
1249985

A Quick Introduction to HTML
621961

1249985
Web pages are written in a text markup language called HTML (HyperText Markup Language).
The idea of HTML is that you annotate, or mark up, regular text with special tags that indicate
structure and formatting. For example, the title of a Web page is defined like this:
1249985

<TITLE>My Home Page</TITLE>

The tags provide general formatting guidelines, but the browsers that display HTML pages have
freedom in how they display things. This keeps the markup simple. The general syntax for HTML
tags is:
1249985

<tag parameters>normal text</tag>

As shown here, the tags usually come in pairs. The open tag may have some parameters, and the
close tag name begins with a slash. The case of a tag is not considered, so <title>, <Title>,
and <TITLE> are all valid and mean the same thing. The corresponding close tag could be </
title>, </Title>, </TITLE>, or even </TiTlE>.
1249985
The <A> tag defines hypertext links that reference other pages on the Web. The hypertext links
connect pages into a Web so that you can move from page to page to page and find related
information. It is the flexibility of the links that makes the Web so interesting. The <A> tag takes
an HREF parameter that defines the destination of the link. If you wanted to link to my home page,
you would put this in your page:
1249985

<A HREF="http://www.beedub.com/">Brent Welch</A>

When this construct appears in a Web page, your browser typically displays "Brent Welch" in blue
underlined text. When you click on that text, your browser switches to the page at the address "http://
www.beedub.com/". There is a lot more to HTML, of course, but this should give you a basic idea
of what is going on in the examples. Table 3-1 summarizes the HTML tags that will be used in the
examples:
1249985

Table 3-1. HTML tags used in the examples

HTML
1249985

Main tag that surrounds the whole document.
1249985

HEAD
1249985

Delimits head section of the HTML document.
1249985

TITLE
1249985

Defines the title of the page.
1249985

BODY
1249985

Delimits the body section. Lets you specify page colors.
1249985

H1 - H6
1249985

HTML defines 6 heading levels: H1, H2, H3, H4, H5, H6.
1249985

P
1249985

Start a new paragraph.
1249985

BR
1249985

One blank line.
1249985

B
1249985

Bold text.
1249985

I
1249985

Italic text.
1249985

A
1249985

Used for hypertext links.
1249985

IMG
1249985

Specify an image.
1249985

DL
1249985

Definition list.
1249985

DT
1249985

Term clause in a definition list.
1249985

DD
1249985

Definition clause in a definition list.
1249985

UL
1249985

An unordered list.
1249985

LI
1249985

A bulleted item within a list.
1249985

TABLE
1249985

Create a table.
1249985

TR
1249985

A table row.
1249985

TD
1249985

A cell within a table row.
1249985

FORM Defines a data entry form.

1249985 1249985

INPUT
1249985

A one-line entry field, checkbox, radio button, or submit button.
1249985

TEXTAREA
1249985

A multiline text field.
1249985

CGI for Dynamic Pages
621961

1249985
There are two classes of pages on the Web: static and dynamic. A static page is written and stored
on a Web server, and the same thing is returned each time a user views the page. This is the easy
way to think about Web pages. You have some information to share, so you compose a page and
tinker with the HTML tags to get the information to look good. If you have a home page, it is
probably in this class.
1249985
In contrast, a dynamic page is computed each time it is viewed. This is how pages that give up-to-
the-minute stock prices work, for example. A dynamic page does not mean it includes animations;
it just means that a program computes the page contents when a user visits the page. The advantage
of this approach is that a user might see something different each time he or she visits the page. As
we shall see, it is also easier to maintain information in a database of some sort and generate the
HTML formatting for the data with a program.
1249985
A CGI (Common Gateway Interface) program is used to compute Web pages. The CGI standard
defines how inputs are passed to the program as well as a way to identify different types of results,
such as images, plain text, or HTML markup. A CGI program simply writes the contents of the
document to its standard output, and the Web server takes care of delivering the document to the
user's Web browser. Example 3-1 is a very simple CGI script:
1249985

Example 3-1. A simple CGI script

puts "Content-Type: text/html"
puts ""
puts "<TITLE>The Current Time</TITLE>"
puts "The time is <B>[clock format [clock seconds]]</B>"

The program computes a simple HTML page that has the current time. Each time a user visits the
page, she will see the current time on the server. The server that has the CGI program and the user
viewing the page might be on different sides of the planet. The output of the program is divided
into two sections: the protocol header and the page contents. In this simple example, the protocol

header just has a Content-Type line that tells your Web browser what kind of data comes next.
A blank line separates the protocol header from the page, which starts with a <TITLE> tag, in this
case.
1249985
The clock command is used twice: once to get the current time in seconds, and a second time to
format the time into a nice-looking string. The clock command is described in detail on page 183.
Fortunately, there is no conflict between the markup syntax used by HTML and the Tcl syntax for
embedded commands, so we can mix the two in the argument to the puts command. Double quotes
are used to group the argument to puts so that the clock command will be executed. Example
3-2 shows what the output of the program will look like:
1249985

Example 3-2. Output of Example 3-1

Content-Type: text/html

<TITLE>The Current Time</TITLE>
The time is <B>Wed Jul 10 14:29:36 2002</B>

This example is a bit sloppy in its use of HTML, but it should display properly in most Web
browsers. Example 3-3 includes all the required tags for a proper HTML document.
1249985

The guestbook.cgi Script
621961

1249985
The guestbook.cgi script computes a page that lists all the registered guests. Example 3-3 is
shown first, and then each part of it is discussed in more detail later. One thing to note right away
is that the HTML tags are generated by procedures that hide the details of the HTML syntax. The
first lines of the script use the UNIX trick to have tclsh interpret the script. This is described on
page 26:
1249985

Example 3-3. The guestbook.cgi script, version 1

#!/bin/sh
# guestbook.cgi
# Implement a simple guestbook page.
# The set of visitors is kept in a simple database.
# The newguest.cgi script will update the database.
# \
exec tclsh "$0" ${1+"$@"}

# The guestbook.data file has the database

# The datafile is in the same directory as the script

set dir [file dirname [info script]]
set datafile [file join $dir guestbook.data]

puts "text/html"
puts ""
set title "Brent's Guestbook"
puts "<HTML><HEAD><TITLE>$title</TITLE></HEAD>"
puts "<BODY BGCOLOR=white TEXT=black>"
puts "<H1>$title</H1>"

if {![file exists $datafile]} {
   puts "No registered guests, yet.
      <P>
      Be the first
      <A href='newguest.html'>registered guest!</A>"
} else {
   puts "The following folks have registered in my GuestBook.
      <P>
      <A href='newguest.html'>Register</A>
      <H2>Guests</H2>"
   catch {source $datafile}
   foreach name [lsort [array names Guestbook]] {
      set item $Guestbook($name)
      set homepage [lindex $item 0]
      set markup [lindex $item 1]
      puts "<H3><A href=$homepage>$name</A></H3>"
      puts $markup
   }
}
puts "</BODY></HTML>"

Using a Script Library File
628024
If you write one CGI script, you are likely to write several. You could start making copies and
modifying your first script, but that quickly becomes hard to maintain. If you learn something new
after writing your third script, will you remember to update the first two scripts you wrote? Probably
not. The best way to approach this problem is to create a collection of Tcl procedures in a file that
you share among all your CGI scripts.
1249985
The Standard Tcl Library, tcllib, provides several packages of procedures that you can use. Later
in this chapter, we will look at the ncgi package that helps handle form data. Before we do that,
let's start a simple collection of our own procedures and learn how to share them among several
different CGI scripts. Suppose you have a file cgihacks.tcl that contains your Tcl procedures.
The source command loads that file into your script. The naive approach shown here probably
won't work:
1249985

source cgihacks.tcl

Loading a file from the same directory as your script
1249985

The problem is that the current directory of the CGI process may not be the same as the directory
that contains the CGI script or the cgihacks.tcl file. You can use the info script
command to find out where the CGI script is, and from that load the supporting file. The file
dirname and file join commands manipulate file names in a platform-independent way.
They are described on page 108. I use the following trick to avoid putting absolute file names into
my scripts, which would have to be changed if the program moves later:
1249985

set dir [file dirname [info script]]
source [file join $dir cgihacks.tcl]

You can also create script libraries as described in Chapter 12. That chapter describes tools to create
an index of procedures so an application can quickly load the procedures it needs, and how to create
packages of procedures so you can keep your code organized. However you set them up, it is always
a good idea to have a library of procedures you share with other applications.
1249985

Beginning the HTML Page
628024
The way you start your HTML page is a great candidate for capturing in a Tcl procedure. For
example, I like to have the page title appear in the TITLE tag in the head, and repeated in an H1
tag at the beginning of the body. You may also have a favorite set of colors or fonts that you want
to specify in the BODY tag. By putting all this into a Tcl procedure, you can make it easy to share
this among all your scripts. If your tastes change tomorrow, then you can change the Tcl procedure
in one spot and affect all CGI scripts that share the procedure. Example 3-4 shows
Cgi_Header that generates a simple standard page header:
1249985

Example 3-4. The Cgi_Header procedure

proc Cgi_Header {title {body {bgcolor=white text=black}}} {
    puts stdout "Content-Type: text/html

<HTML>
<HEAD>
<TITLE>$title</TITLE>
</HEAD>
<BODY $body>
<H1>$title</H1>"
}

The Cgi_Header procedure takes as arguments the title for the page and some optional parameters
for the HTML BODY tag. The procedure definition uses the syntax for an optional parameter, so
you do not have to pass bodyparams to Cgi_Header. The default specifies black text on a
white background to avoid the standard gray background of most browsers. Default values for
procedure parameters are described on page 87.
1249985

Example 3-5. The guestbook.cgi script, version 2

#!/bin/sh
# guestbook.cgi
# Implement a simple guestbook page.
# The set of visitors is kept in a simple database.
# The newguest.cgi script will update the database.
# \
exec tclsh "$0" ${1+"$@"}

# The guestbook.data file has the database
# The datafile is in the same directory as the script

set dir [file dirname [info script]]
set datafile [file join $dir guestbook.data]

# Load our supporting Tcl procedures to define Cgi_Header

source [file join $dir cgihacks.tcl]
Cgi_Header "Brent's Guestbook"

if {![file exists $datafile]} {
   puts "No registered guests, yet.
      <P>
      Be the first
      <A href='newguest.html'>registered guest!</A>"
} else {
   puts "The following folks have registered in my GuestBook.
      <P>
      <A href='newguest.html'>Register</A>
      <h2>Guests</h2>"
   catch {source $datafile}
   foreach name [lsort [array names Guestbook]] {
      set item $Guestbook($name)
      set homepage [lindex $item 0]
      set markup [lindex $item 1]
      puts "<H3><A href=$homepage>$name</A></H3>"
      puts $markup
   }
}
puts "</BODY></HTML>"

Example 3-5 is a new version of the original CGI script that loads the cgihacks.tcl file and
uses Cgi_Header. The Cgi_Header procedure just contains a single puts command that
generates the standard boilerplate that appears at the beginning of the output. Note that several lines
are grouped together with double quotes. Double quotes are used so that the variable references
mixed into the HTML are substituted properly. The output of the Cgi_Header procedure matches
what we wrote by hand in Example 3-3.
1249985

Sample Output of the CGI Script
628024
The program tests to see whether there are any registered guests or not. The file command, which
is described in detail on page 108, is used to see whether there is any data. The exclamation point
means "not" in a boolean expression:
1249985

if {![file exists $datafile]} {

If the database file does not exist, a different page is displayed to encourage a registration. The page
includes a hypertext link to a registration page, newguest.html, which is described on page 43.
The output of the program would be as below in Example 3-6 if there were no data file:
1249985

Example 3-6. Initial output of guestbook.cgi with no data

Content-Type: text/html

<HTML>
<HEAD>
<TITLE>Brent's Guestbook</TITLE>
</HEAD>
<BODY BGCOLOR=white TEXT=black>
<H1>Brent's Guestbook</H1>
<P>
No registered guests.
   <P>
   Be the first
   <A HREF="newguest.html">registered guest!</A>
</BODY></HTML>

Note the inconsistent indentation of the HTML that comes from the indentation in the puts
command used for that part of the page. The browser doesn't care about white space in the HTML.
You have a choice between lining up the Tcl commands in your CGI script, or lining up the HTML
output. Here we have two different examples. The Cgi_Header procedure produces output that
is lined up, but the procedure definition looks a bit odd. The main script, in contrast, keeps its Tcl
commands neatly indented, but that shows up in the output. If you generate most of your HTML
from code, you may choose to keep your code tidy.
1249985
Example 3-7 shows the output of the guestbook.cgi script when there is some data in the data
file:
1249985

Example 3-7. Output of guestbook.cgi with guestbook data

Content-Type: text/html

<HTML>
<HEAD>

<TITLE>Brent's Guestbook</TITLE>
</HEAD>
<BODY BGCOLOR=white TEXT=black>
<H1>Brent's Guestbook</H1>
<P>
The following folks have registered in my guestbook.
   <P>
   <A HREF='newguest.html'>Register</A>
   <H2>Guests</H2>
<H3><A HREF="http://www.beedub.com/">Brent Welch</A></H3>
<IMG SRC="http://www.beedub.com/welch.gif">
</BODY></HTML>

Using a Tcl Array for the Database
628024
The data file contains Tcl commands that define an array that holds the guestbook data. If this file
is kept in the same directory as the guestbook.cgi script, then you can compute its name:
1249985

set dir [file dirname [info script]]
set datafile [file join $dir guestbook.data]

By using Tcl commands to represent the data, we can load the data with the source command.
The catch command is used to protect the script from a bad data file, which will show up as an
error from the source command. Catching errors is described in detail on page 85:
1249985

catch {source $datafile}

The Guestbook variable is the array defined in guestbook.data. Array variables are the
topic of Chapter 8. Each element of the array is defined with a Tcl command that looks like this:
1249985

set Guestbook(key) {url markup}

The person's name is the array index, or key. The value of the array element is a Tcl list with two
elements: their URL and some additional HTML markup that they can include in the guestbook.
Tcl lists are the topic of Chapter 5. The following example shows what the command looks like
with real data:
1249985

set {Guestbook(Brent Welch)} {
    http://www.beedub.com/
    {<img src=http://www.beedub.com/welch.gif>}
}

The spaces in the name result in additional braces to group the whole variable name and each list
element. This syntax is explained on page 96. Do not worry about it now. We will see on page 46
that all the braces in the previous statement are generated automatically. The main point is that the
person's name is the key, and the value is a list with two elements.

1249985
The array names command returns all the indices, or keys, in the array, and the lsort
command sorts these alphabetically. The foreach command loops over the sorted list, setting the
loop variable x to each key in turn:
1249985

foreach name [lsort [array names Guestbook]] {

The lsort command will sort the names based on the person's first name. You can have
lsort sort things in a variety of ways. One trick we can use here is to have lsort treat each key
as a list and sort on the last item in the list (i.e., the last name):
1249985

foreach name [lsort -index end [array names Guestbook]] {

The lsort command is described in more detail on page 70. The foreach command assigns
name to each key of the Guestbook array. We get the value like this:
1249985

set item $Guestbook($name)

The two list elements are extracted with lindex, which is described on page 68.
1249985

set homepage [lindex $item 0]
set markup [lindex $item 1]

We generate the HTML for the guestbook entry as a level-three header that contains a hypertext
link to the guest's home page. We follow the link with any HTML markup text that the guest has
supplied to embellish his or her entry:
1249985

puts "<H3><a href=$homepage>$name</a></H3>"
puts $markup

The homepage and markup variables are not strictly necessary, and the code could be written
more compactly without them. However, the variables make the code more understandable. Here
is what it looks like without the temporary variables:
1249985

puts "<H3><a href=[lindex $item 0]>$name</a></H3>"
puts [lindex $item 1]

Defining Forms and Processing Form Data
621961

1249985

The guestbook.cgi script only generates output. The other half of CGI deals with input from
the user. Input is more complex for two reasons. First, we have to define another HTML page that
has a form for the user to fill out. Second, the data from the form is organized and encoded in a
standard form that must be decoded by the script. Example 3-8 on page 43 defines a very simple
form, and the procedure that decodes the form data is shown in Example 11-6 on page 165.
1249985
The guestbook page contains a link to newguest.html. This page contains a form that lets a
user register his or her name, home page URL, and some additional HTML markup. The form has
a submit button. When a user clicks that button in her browser, the information from the form is
passed to the newguest.cgi script. This script updates the database and computes another page
for the user that acknowledges the user's contribution.
1249985

The newguest.html Form
628024
An HTML form contains tags that define data entry fields, buttons, checkboxes, and other elements
that let the user specify values. For example, a one-line entry field that is used to enter the home
page URL is defined like this:
1249985

<INPUT TYPE=text NAME=url>

The INPUT tag is used to define several kinds of input elements, and its type parameter indicates
what kind. In this case, TYPE=text creates a one-line text entry field. The submit button is defined
with an INPUT tag that has TYPE=submit, and the VALUE parameter becomes the text that
appears on the submit button:
1249985

<INPUT TYPE=submit NAME=submit VALUE=Register>

A general type-in window is defined with the TEXTAREA tag. This creates a multiline, scrolling
text field that is useful for specifying lots of information, such as a free-form comment. In our case,
we will let guests type in HTML that will appear with their guestbook entry. The text between the
open and close TEXT-AREA tags is inserted into the type-in window when the page is first
displayed.
1249985

<TEXTAREA NAME=markup ROWS=10 COLS=50>Hello.</TEXTAREA>

A common parameter to the form tags is NAME=something. This name identifies the data that
will come back from the form. The tags also have parameters that affect their display, such as the
label on the submit button and the size of the text area. Those details are not important for our
example. The complete form is shown in Example 3-8:
1249985

Example 3-8. The newguest.html form

<HTML>
<HEAD>
<TITLE>Register in my Guestbook</TITLE>
</HEAD>
<BODY BGCOLOR=white TEXT=black>

<FORM ACTION="newguest.cgi" METHOD="POST">

<H1>Register in my Guestbook</H1>
<UL>
<LI>Name <INPUT TYPE="text" NAME="name" SIZE="40">
<LI>URL  <INPUT TYPE="text" NAME="url" SIZE="40">
<P>
If you don't have a home page, you can use an email URL like "mailto:welch@acm.org"
<LI>Additional HTML to include after your link:
<BR>

<TEXTAREA NAME="html" COLS="60" ROWS="15">
</TEXTAREA>
<LI><INPUT TYPE="submit" NAME="new" VALUE="Add me to your guestbook">
<LI><INPUT TYPE="submit" NAME="update" VALUE="Update my guestbook entry">
</UL>
</FORM>

</BODY>
</HTML>

The ncgi and cgi.tcl Packages
628024
The newguest.cgi script uses the ncgi package to process form data. This is one of many
packages available in the Standard Tcl Library, commonly known as "tcllib". If you don't have
tcllib installed, you can find it on the CD-ROM, on SourceForge at www.sf.net/projects/tcllib, or
via the main www.tcl.tk Web site. If your Tcl installation includes tcllib, then you use the
package command to load the package.
1249985

package require ncgi

The procedures in the ncgi package are in the ncgi namespace. Tcl namespaces are described
in detail in Chapter 14. Procedures in a namespace are qualified with the name of the namespace
and :: syntax. For example, the standard setup procedure for a CGI script is ncgi::parse.
1249985
The "n" in ncgi is for "new". Don Libes wrote the original package for CGI scripts known as
cgi.tcl. There is also the cgilib.tcl package that contains Cgi_Header and some other
procedures described in earlier editions of this book. The ncgi and html packages of tcllib provide
most of the features in both cgi.tcl and cgilib.tcl, but follow the standard namespace
conventions use by the packages in tcllib. You can still find cgi.tcl on the Web at
1249985

http://expect.nist.gov/cgi.tcl/

The newguest.cgi Script
628024
When the user clicks the Submit button in her browser, the data from the form is passed to the
program identified by the ACTION parameter of the form tag. That program takes the data, does
something useful with it, and then returns a new page for the browser to display. In our case, the
FORM tag names newguest.cgi as the program to handle the data:
1249985

<FORM ACTION=newguest.cgi METHOD=POST>

The CGI specification defines how the data from the form is passed to the program. The data is
encoded and organized so that the program can figure out the values the user specified for each
form element. The encoding is handled rather nicely with some regular expression tricks that are
done in ncgi::parse. ncgi::parse saves the form data, and ncgi::value gets a form
value in the script. These procedures are described in Example 11-6 on page 165. Example 3-9
starts out by calling ncgi::parse:
1249985

Example 3-9. The newguest.cgi script

#!/bin/sh
# \
exec tclsh "$0" ${1+"$@"}

# Use the ncgi package from tcllib to process form data

package require ncgi
ncgi::parse

# Load our data file and supporting procedures

set dir [file dirname [info script]]
set datafile [file join $dir guestbook.data]
source [file join $dir cgihacks.tcl]

# Open the datafile in append mode

if {[catch {open $datafile a} out]} {
   Cgi_Header "Guestbook Registration Error" \
      {BGCOLOR=black TEXT=red}
   puts "<P>Cannot open the data file<P>"
   puts $out;# the error message
   exit 0
}

# Append a Tcl set command that defines the guest's entry

puts $out ""
puts $out [list set Guestbook([ncgi::value name]) \
   [list [ncgi::value url] [ncgi::value html]]]
close $out

# Return a page to the browser

Cgi_Header "Guestbook Registration Confirmed" \
   {BGCOLOR=white TEXT=black}

puts "
<TABLE BORDER=1>
<TR><TD>Name</TD>
<TD>[ncgi::value name]</TD></TR>
<TR><TD>URL</TD>
<TD><A HREF='[ncgi::value url]'>[ncgi::value url]</A></TD></TR>
<TR><TD>Extra HTML</TD>
<TD>[ncgi::value html]</TD></TR>
</TABLE>
"

puts </BODY></HTML>

Using Tcl Scripts to Store Data
628024
The main idea of the newguest.cgi script is that it saves the data to a file as a Tcl command
that defines an element of the Guestbook array. This lets the guestbook.cgi script simply
load the data by using the Tcl source command. This trick of storing data as a Tcl script saves
us from the chore of defining a new file format and writing code to parse it. Instead, we can rely
on the well-tuned Tcl implementation to do the hard work for us efficiently.
1249985
The script opens the datafile in append mode so that it can add a new record to the end. Opening
files is described in detail on page 116. The script uses a catch command to guard against errors.
If an error occurs, a page explaining the error is returned to the user. Working with files is one of
the most common sources of errors (permission denied, disk full, file-not-found, and so on), so I
always open the file inside a catch statement:
1249985

if {[catch {open $datafile a} out]} {
    # an error occurred
} else {
    # open was ok
}

In this command, the variable out gets the result of the open command, which is either a file
descriptor or an error message. This style of using catch is described in detail in Example 6-14 on
page 83.
1249985

Use list to generate Tcl commands.
1249985

The script writes the data as a Tcl set command. The list command is used to format the data
properly:
1249985

puts $out [list set Guestbook([ncgi::value name]) \
     [list [ncgi::value url] [ncgi::value html]]]

There are two lists. First, the url and html values are formatted into one list. This list will be the
value of the array element. Then the whole Tcl command is formed as a list. In simplified form,
the command is generated from this:
1249985

list set variable value
            

Using the list command ensures that the result will always be a valid Tcl command that sets the
variable to the given value. This is a very important technique. If you want to generate Tcl
commands, the best way to do it is to generate lists using list manipulation commands. The
list command is described in more detail on page 65.
1249985

Handling Errors in CGI Scripts
621961
One of the more frustrating aspects of CGI programming is that errors in your script result in blank
browser pages, and it may be difficult or impossible to find any trace of the error message. The
other main problem is that your Web server may not be configured properly to find your CGI script.
I use two simple tricks to track down the source of these errors. The first trick simply verifies that
my script has run at all by creating an empty file somewhere on the Web server. On a UNIX system,
you can put this line at the beginning of your script:
1249985

close [open /tmp/my_cgi_script_ran w]

When you aim the browser at your CGI script, it should at least create the file. If not, then the Web
server cannot find your script, or it cannot find the Tclsh required by your script. Double-check
your setup and the #! line in your script. On Windows, your best bet may be to use the TclHttpd
Web server, which has a built-in ability to run Tcl CGI scripts. TclHttpd has other even cooler ways
to generate pages, too.
1249985
If your script suddenly stops working after you've modified it, then you have introduced a
programming bug. I generally put all of the script into a catch statement and print out any errors
that occur. That way the errors will be displayed by the browser instead of filed into the void by
your Web server. Example 3-10 shows the newguest.cgi script rewritten so the catch

statement surrounds all the statements. At the end, the value of the errorInfo variable is printed
out if an error has occurred:
1249985

Example 3-10. The newguest.cgi script with error handling

#!/bin/sh
# \
exec tclsh "$0" ${1+"$@"}

# Trap all errors

if {[catch {

# Use the ncgi package from tcllib to process form data

package require ncgi
ncgi::parse

# Load our data file and supporting procedures

set dir [file dirname [info script]]
set datafile [file join $dir guestbook.data]
source [file join $dir cgihacks.tcl]

# Open the datafile in append mode

set out [open $datafile a]

# Append a Tcl set command that defines the guest's entry

puts $out ""
puts $out [list set Guestbook([ncgi::value name]) \
   [list [ncgi::value url] [ncgi::value html]]]
close $out

# Return a page to the browser
Cgi_Header "Guestbook Registration Confirmed" \
    {BGCOLOR=white TEXT=black}

puts "
<TABLE BORDER=1>
<TR><TD>Name</TD>
<TD>[ncgi::value name]</TD></TR>
<TR><TD>URL</TD>
<TD><A HREF='[ncgi::value url]'>[ncgi::value url]</A></TD></TR>
<TR><TD>Extra HTML</TD>
<TD>[ncgi::value html]</TD></TR>
</TABLE>
</BODY></HTML>
"

# End of main script

} err]} {

   # Error occurred - display in the Web page

   puts "Content-Type: text/plain"
   puts ""
   puts "CGI error occurred in [info script]"
   puts $errorInfo

}

Next Steps
621961
There are a number of details that can be added to this example. Users may want to update their
entry, for example. They could do that now, but they would have to retype everything. They might
also like a chance to check the results of their registration and make changes before committing
them. This requires another page that displays their guest entry as it would appear on a page, and
also has the fields that let them update the data.
1249985
The details of how a CGI script is hooked up with a Web server vary from server to server. You
should ask your local Webmaster for help if you want to try this out on your local Web site. The
Tcl Web Server comes with this guestbook example already set up, plus it has a number of other
very interesting ways to generate pages. My own taste in Web page generation has shifted from
CGI to a template-based approach supported by the Tcl Web Server. This is the topic of Chapter
18.
1249985
The next few chapters describe basic Tcl commands and data structures. We return to the CGI
example in Chapter 11 on regular expressions.
1249985

