
Chapter 8. Tcl Arrays
484

This chapter describes Tcl arrays, which provide a flexible mechanism to build many other
data structures in Tcl. Tcl command described is: array.
1249985
An array is a Tcl variable with a string-valued index. You can think of the index as a key, and
the array as a collection of related data items identified by different keys. The index, or key,
can be any string value. Internally, an array is implemented with a hash table, so the cost of
accessing each array element is about the same. Before Tcl 8.0, arrays had a performance
advantage over lists that took time to access proportional to the size of the list.
1249985
The flexibility of arrays makes them an important tool for the Tcl programmer. A common
use of arrays is to manage a collection of variables, much as you use a C struct or Pascal record.
This chapter shows how to create several simple data structures using Tcl arrays.
1249985

Array Syntax
621961

1249985
The index of an array is delimited by parentheses. The index can have any string value, and
it can be the result of variable or command substitution. Array elements are defined with
set:
1249985

set arr(index) value
            

The value of an array element is obtained with $ substitution:
1249985

set foo $arr(index)

Example 8-1 uses the loop variable value $i as an array index. It sets arr(x) to the product
of 1 * 2 * ... * x:
1249985

Example 8-1. Using arrays

set arr(0) 1
for {set i 1} {$i <= 10} {incr i} {
   set arr($i) [expr {$i * $arr([expr {$i-1}])}]
}

Complex Indices
628024
An array index can be any string, like orange, 5, 3.1415, or foo,bar. The examples in this
chapter, and in this book, often use indices that are pretty complex strings to create flexible
data structures. As a rule of thumb, you can use any string for an index, but avoid using a
string that contains spaces.
1249985

Parentheses are not a grouping mechanism.
1249985

The main Tcl parser does not know about array syntax. All the rules about grouping and
substitution described in Chapter 1 are still the same in spite of the array syntax described
here. Parentheses do not group like curly braces or quotes, which is why a space causes
problems. If you have complex indices, use a comma to separate different parts of the index.
If you use a space in an index instead, then you have a quoting problem. The space in the
index needs to be quoted with a backslash, or the whole variable reference needs to be
grouped:
1249985

set {arr(I'm asking for trouble)} {I told you so.}
set arr(I'm\ asking\ for\ trouble) {I told you so.}

If the array index is stored in a variable, then there is no problem with spaces in the variable's
value. The following works well:
1249985

set index {I'm asking for trouble}
set arr($index) {I told you so.}

Array Variables
628024
You can use an array element as you would a simple variable. For example, you can test for
its existence with info exists, increment its value with incr, and append elements to
it with lappend:
1249985

if {[info exists stats($event)]} {incr stats($event)}

You can delete an entire array, or just a single array element with unset. Using unset on
an array is a convenient way to clear out a big data structure.
1249985
It is an error to use a variable as both an array and a normal variable. The following is an error:
1249985

set arr(0) 1
set arr 3
=> can't set "arr": variable is array

The name of the array can be the result of a substitution. This is a tricky situation, as shown
in Example 8-2:
1249985

Example 8-2. Referencing an array indirectly

set name TheArray
=> TheArray
set ${name}(xyz) {some value}
=> some value
set x $TheArray(xyz)
=> some value
set x ${name}(xyz)
=> TheArray(xyz)
set x [set ${name}(xyz)]
=> some value
                        

A better way to deal with this situation is to use the upvar command, which is introduced
on page 91. The previous example is much cleaner when upvar is used:
1249985

Example 8-3. Referencing an array indirectly using upvar

set name TheArray
=> TheArray
                            upvar 0 $name a
set a(xyz) {some value}
=> some value
set x $TheArray(xyz)
=> some value

The array Command
621961

1249985
The array command returns information about array variables. The array names
command returns the index names that are defined in the array. If the array variable is not
defined, then array names just returns an empty list. It allows easy iteration through an
array with a foreach loop:
1249985

foreach index [array names arr pattern] {
    # use arr($index)
}

The order of the names returned by array names is arbitrary. It is essentially determined
by the hash table implementation of the array. You can limit what names are returned by
specifying a pattern that matches indices. The pattern is the kind supported by the string
match command, which is described on page 53.
1249985
It is also possible to iterate through the elements of an array one at a time using the search-
related commands listed in Table 8-1. The ordering is also random, and I find the foreach
over the results of array names much more convenient. If your array has an extremely
large number of elements, or if you need to manage an iteration over a long period of time,
then the array search operations might be more appropriate. Frankly, I never use them. Table
8-1 summarizes the array command:
1249985

Table 8-1. The array command

array exists arr
1249985

Returns 1 if arr is an array variable.
1249985

array get arr ?pattern?
1249985

Returns a list that alternates between an index and the
corresponding array value. pattern selects matching indices.
If not specified, all indices and values are returned.
1249985

array names arr ?mode? ?pattern?
1249985

Returns the list of all indices defined for arr, or those that
match pattern. mode specifies the pattern type and may be
-exact, -glob (default) or -regexp.
1249985

array set arr list
1249985

Initializes the array arr from list, which has the same form
as the list returned byarray get.

1249985

array size arr
1249985

Returns the number of indices defined for arr.
1249985

array unset arr ?pattern?
1249985

Unset elements in arr matching the specified glob-style
pattern. If not specified, unset arr. (Tcl 8.3)
1249985

array startsearch arr
1249985

Returns a search token for a search through arr.
1249985

array nextelement arr id
1249985

Returns the value of the next element in arr in the search
identified by the token id. Returns an empty string if no more
elements remain in the search.
1249985

array anymore arr id
1249985

Returns 1 if more elements remain in the search.
1249985

array donesearch arr id
1249985

Ends the search identified by id.
1249985

array statistics arr
1249985

Returns statistics about the array hash table. (Tcl 8.4)
1249985

Converting Between Arrays and Lists
628024
The array get and array set operations are used to convert between an array and a
list. The list returned by array get has an even number of elements. The first element is
an index, and the next is the corresponding array value. The list elements continue to
alternate between index and value. The list argument to array set must have the same
structure.
1249985

array set fruit {
   best   kiwi
   worst  peach
   ok     banana
}
array get fruit
=> ok banana best kiwi worst peach
            

Another way to loop through the contents of an array is to use array get and the two-
variable form of the foreach command.
1249985

foreach {key value} [array get fruit] {
   # key is ok, best, or worst

   # value is some fruit
}

Passing Arrays by Name
628024
The upvar command works on arrays. You can pass an array name to a procedure and use
the upvar command to get an indirect reference to the array variable in the caller's scope.
This is illustrated in Example 8-4, which inverts an array. As with array names, you can
specify a pattern to array get to limit what part of the array is returned. This example uses
upvar because the array names are passed into the ArrayInvert procedure. The inverse
array does not need to exist before you call ArrayInvert.
1249985

Example 8-4. ArrayInvert inverts an array

proc ArrayInvert {arrName inverseName {pattern *}} {
   upvar $arrName array $inverseName inverse
   foreach {index value} [array get array $pattern] {
      set inverse($value) $index
   }
}

Building Data Structures with Arrays
621961

1249985
This section describes several data structures you can build with Tcl arrays. These examples
are presented as procedures that implement access functions to the data structure. Wrapping
up your data structures in procedures is good practice. It shields the user of your data
structure from the details of its implementation.
1249985

Use arrays to collect related variables.
1249985

A good use for arrays is to collect together a set of related variables for a module, much as
one would use a record in other languages. By collecting these together in an array that has
the same name as the module, name conflicts between different modules are avoided. Also,
in each of the module's procedures, a single global statement will suffice to make all the

state variables visible. You can also use upvar to manage a collection of arrays, as shown in
Example 8-9 on page 101.
1249985

Simple Records
628024
Suppose we have a database of information about people. The following examples show
three different ways to store the employee name, ID, manager, and phone number. Each
example implements Emp_AddRecord that stores the values, and one example accessor
function that returns information about the employee (e.g., Emp_Manager.) By using simple
procedures to return fields of the record, the implementation is hidden so that you can
change it more easily. Example 8-5 uses on array for each field. The name of the person is the
index into each array:
1249985

Example 8-5. Using arrays for records, version 1

proc Emp_AddRecord {id name manager phone} {
   global employeeID employeeManager \
      employeePhone employeeName
   set employeeID($name) $id
   set employeeManager($name) $manager
   set employeePhone($name) $phone
   set employeeName($id) $name
}
proc Emp_Manager {name} {
   global employeeManager
   return $employeeManager($name)
}

The employeeName array provides a secondary key. It maps from the employee ID to the
name so that the other information can be obtained if you have an ID instead of a name.
Example 8-6 implements the same little database using a single array with more complex
indices:
1249985

Example 8-6. Using arrays for records, version 2

proc Emp_AddRecord {id name manager phone} {
   global employee
   set employee(id,$name) $id
   set employee(manager,$name) $manager
   set employee(phone,$name) $phone
   set employee(name,$id) $name
}
proc Emp_Manager {name} {
   global employee
   return $employee(manager,$name)
}

Example 8-7 shows the last approach. Each array element is a list of fields, and the accessor
functions hide the lindex command used to pick out the right field. Here the cross
referencing by ID is implement differently. If we can assume that names and IDs are distinct,
we can keep the cross reference in the same array:
1249985

Example 8-7. Using arrays for records, version 3

proc Emp_AddRecord {id name manager phone} {
   global employee
   set employee($name) [list $name $id $manager $phone]
   set employee($id) $name
}
proc Emp_Manager {name} {
   global employee
   return [lindex $employee($name) 2]
}

The difference between these three approaches is partly a matter of taste. Using a single
array can be more convenient because there are fewer variables to manage. Using the lists
for the fields is probably the most space efficient because there are fewer elements in the
array, but maintaining the lindex offsets is tedious. In any case, you should hide the
implementation in a small set of procedures.
1249985

A Stack
628024
A stack can be implemented with either a list or an array. If you use a list, then the push and
pop operations have a runtime cost that is proportional to the size of the stack. If the stack
has a few elements this is fine. If there are a lot of items in a stack, you may wish to use arrays
instead.
1249985

Example 8-8. Using a list to implement a stack

proc Push { stack value } {
   upvar $stack list
   lappend list $value
}
proc Pop { stack } {
   upvar $stack list
   set value [lindex $list end]
   set list [lrange $list 0 [expr [llength $list]-2]]
   return $value
}

In these examples, the name of the stack is a parameter, and upvar is used to convert that
into the data used for the stack. The variable is a list in Example 8-8 and an array in Example
8-9. The user of the stack module does not have to know.
1249985

The array implementation of a stack uses one array element to record the number of items
in the stack. The other elements of the array have the stack values. The Push and Pop
procedures both guard against a nonexistent array with the info exists command. When
the first assignment to S(top) is done by Push, the array variable is created in the caller's
scope. The example uses array indices in two ways. The top index records the depth of the
stack. The other indices are numbers, so the construct $S($S(top)) is used to reference
the top of the stack.
1249985

Example 8-9. Using an array to implement a stack

proc Push { stack value } {
   upvar $stack S
   if {![info exists S(top)]} {
      set S(top) 0
   }
   set S($S(top)) $value
   incr S(top)
}
proc Pop { stack } {
   upvar $stack S
   if {![info exists S(top)]} {
      return {}
   }
   if {$S(top) == 0} {
      return {}
   } else {
      incr S(top) -1
      set x $S($S(top))
      unset S($S(top))
      return $x
   }
}

A List of Arrays
628024
Suppose you have many arrays, each of which stores some data, and you want to maintain
an overall ordering among the data sets. One approach is to keep a Tcl list with the name of
each array in order. Example 8-10 defines RecordInsert to add an array to the list, and an
iterator function, RecordIterate, that applies a script to each array in order. The iterator
uses upvar to make data an alias for the current array. The script is executed with eval,
which is described in detail in Chapter 10. The Tcl commands in script can reference the
arrays with the name data:
1249985

Example 8-10. A list of arrays

proc RecordAppend {listName arrayName} {
   upvar $listName list
   lappend list $arrayName

}
proc RecordIterate {listName script} {
   upvar $listName list
   foreach arrayName $list {
      upvar #0 $arrayName data
      eval $script
   }
}

Another way to implement this list-of-records structure is to keep references to the arrays
that come before and after each record. Example 8-11 shows the insert function and the
iterator function when using this approach. Once again, upvar is used to set up data as an
alias for the current array in the iterator. In this case, the loop is terminated by testing for the
existence of the next array. It is perfectly all right to make an alias with upvar to a nonexistent
variable. It is also all right to change the target of the upvar alias. One detail that is missing
from the example is the initialization of the very first record so that its next element is the
empty string:
1249985

Example 8-11. A list of arrays

proc RecordInsert {recName afterThis} {
   upvar $recName record $afterThis after
   set record(next) $after(next)
   set after(next) $recName
}
proc RecordIterate {firstRecord body} {
   upvar #0 $firstRecord data
   while {[info exists data]} {
      eval $body
      upvar #0 $data(next) data
   }
}

A Simple In-Memory Database
628024
Suppose you have to manage a lot of records, each of which contain a large chunk of data
and one or more key values you use to look up those values. The procedure to add a record
is called like this:
1249985

Db_Insert keylist datablob
            

The datablob might be a name, value list suitable for passing to array set, or simply a
large chunk of text or binary data. One implementation of Db_Insert might just be:
1249985

foreach key $keylist {
    lappend Db($key) $datablob
}

The problem with this approach is that it duplicates the data chunks under each key. A better
approach is to use two arrays. One stores all the data chunks under a simple ID that is
generated automatically. The other array stores the association between the keys and the
data chunks. Example 8-12, which uses the namespace syntax described in Chapter 14,
illustrates this approach. The example also shows how you can easily dump data structures
by writing array set commands to a file, and then load them later with a source
command:
1249985

Example 8-12. A simple in-memory database

namespace eval db {
   variable data       ;# Array of data blobs
   variable uid 0      ;# Index into data
   variable index      ;# Cross references into data
}
proc db::insert {keylist datablob} {
   variable data
   variable uid
   variable index
   set data([incr uid]) $datablob
   foreach key $keylist {
      lappend index($key) $uid
   }
}
proc db::get {key} {
   variable data
   variable index
   set result {}
   if {![info exist index($key)]} {
      return {}
   }
   foreach uid $index($key) {
      lappend result $data($uid)
   }
   return $result
}
proc db::save {filename} {
   variable uid
   set out [open $filename w]
   puts $out [list namespace eval db \
      [list variable uid $uid]]
   puts $out [list array set db::data [array get db::data]]
   puts $out [list array set db::index [array get db::index]]
   close $out
}
proc db::load {filename} {
   source $filename
}

Alternatives to Using Arrays
628024
While Tcl arrays are flexible and general purpose, they are not always the best solution to
your data structure problems. If you find yourself building elaborate data structures, you

should consider implementing a C library to encapsulate the data structure and expose it to
the scripting level with Tcl commands. For example, Chapter 47 implements a blob data
structure in C. You can also use the SWIG code generator can quickly generate a Tcl command
interface for a C API. Find out about SWIG at http://www.swig.org.
1249985
The Metakit embedded database provides an efficient, easy, scriptable database for Tcl. It is
more powerful than the simple "flat file" databases implemented in this Chapter, but it is not
a full SQL database. It is part of Tclkit, or you can use it with the mk4tcl extension. Tclkit and
Metakit are described in Chapter 22.
1249985

